Numerical Integration Using SCILAB

By

Gilberto E. Urroz, Ph.D., P.E.

Distributed by

*i*nfoClearinghouse.com

A "zip" file containing all of the programs in this document (and other SCILAB documents at InfoClearinghouse.com) can be downloaded at the following site:

http://www.engineering.usu.edu/cee/faculty/gurro/Software_Calculators/Scilab_Docs/ScilabBookFunctions.zip

The author's SCILAB web page can be accessed at:

http://www.engineering.usu.edu/cee/faculty/gurro/Scilab.html

Please report any errors in this document to: gurro@cc.usu.edu

Numerical integration using SCILAB

Integrals can be interpreted as the area under the curve of the function f(x) in a given interval a < x < b. Such an integral is written as

$$\int_a^b f(x)dx$$

where the term *dx*, referred to as the differential of x, indicates the variable of integration. In the next section, we present methods to estimate the value of an integral by using summations.

Integrals calculated through summation of rectangles

The integral of a function *f(x)* in an interval *(a,b),* is defined as the limit of the sum

$$S_n = \sum_{i=1}^{n} f(\xi_i)\Delta x_i,$$

as $\Delta x_i \rightarrow 0$, or $n \rightarrow \infty$. The values Δx_i represent the length of *n* sub-intervals in (*a,b*), so that the values ξ_i are contained within the *i*-th sub-interval, i.e., $x_i \le \xi_i \le x_{i+1}$. The sub-intervals are limited by the values $x_1, x_2, \ldots, x_n, x_{n+1}$, therefore, $\Delta x_i = x_{i+1} - x_i$,

The figure below illustrates the meaning of the terms in the summation. The terms $f(\xi_i)\Delta x_i$ represent increments of area, ΔA_i, under the curve $y = f(x)$ in the interval (*a,b*).



While there are no restrictions in the way we may divide the interval (*a,b*) to generate n sub-intervals, or where to select ξ_i within a sub-interval, dividing it into n equally-spaced sub-intervals, and selecting the values of ξ_i in a regular fashion, facilitates the calculation of the summation.

To divide the interval (*a*,*b*) into *n* sub-intervals we take,

$$\Delta x = (b-a)/n,$$

therefore,

$$x_1 = a,\ x_2 = x_1 + \Delta x,\ x_3 = x_1 + 2\cdot\Delta x,\ \ldots,\ x_i = x_1 + (i-1)\cdot\Delta x,\ \ldots,\ x_n = x_1 + (n-1)\cdot\Delta x = b.$$

The value of can be selected to be the leftmost value in the sub-interval (x_i, x_{i+1}), i.e., $\xi_i = x_i$, the center of the sub-interval, i.e., $\xi_i = (x_i + x_{i+1})/2$, or the right-most value of the sub-interval, i.e., = x_{i+1}. Suppose that we call SL_n the summation when $\xi_i = x_i$, then we can write:

$$SL_n = \left(\frac{b-a}{n}\right)\cdot\sum_{i=1}^{n} f(x_i).$$

If we call SM_n the summation when $\xi_i = (x_i + x_{i+1})/2$, then we have

$$SM_n = \left(\frac{b-a}{n}\right)\cdot\sum_{i=1}^{n} f\left(\frac{x_i + x_{i+1}}{2}\right)$$

Finally, for the summation when $\xi_i = x_{i+1}$, we have

$$SU_n = \left(\frac{b-a}{n}\right)\cdot\sum_{i=1}^{n} f(x_{i+1}).$$

The following function, *Sumint* (*Sum*mation as *int*egrals), can be used to calculate the summations SL_n, SM_n, and SU_n, as defined above, given the values of the integration limits, *a* and *b*, the number of sub-intervals for the summation, *n*, and the function to be integrated, *f*. The general call to the function is

[I] = Sumint(sum_type,a,b,n,f)

where *sum_type* can take the values *'L'* for SL_n, *'M'* for SM_n, and *'U'* for SU_n. The function returns the value of the summation requested as well as a graph showing the function *y = f(x)* in the interval *a<x<b* and the rectangles whose area represent the summation. The graph for SL_n is shown in SCILAB graphics window 1, that for SM_n is shown in SCILAB graphics window 2, and that for SU_n is shown in SCILAB graphics window 3.

A listing of function *Sumint* follows:

```
function [I] = Sumint(stype,a,b,n,f)

//Calculates the summation corresponding
//to the integral:
//I = integral from a to b of f(x) dx
//
//The variable 'stype' can be one of the
//following:
//
```



```
//  'L' - lower sum
//  'M' - middle sum
//  'U' - upper sum
//
//n is the number of sub-intervals in [a,b]

//Checking that the proper value of 'stype' is used:
if (stype<>'L') & (stype<>'M') & (stype<>'U') then
       error('Sumint - stype must be L, M, or U between brackets');
       abort;
end;

//Calculating parameters for sum and plot
Dx = (b-a)/(n-1);      //Increment in x
x = [a:Dx:b];          //List of values of x
y = feval(x,f);        //List of values of the function, y = f(x)
[n m] = size(x);       //Size of vectors x and y
xmin = min(x);         //Minimum value of x
xmax = max(x);         //Minimum value of y
ym = f(xmax);          //Value of y = f(xmax)
yy = [y(2:m), ym];     //Vector of values of y shifted one Dx to the right
ymin = min(y);         //Minimum value of y
ymax = max(y);         //Maximum value of y

//Change ymin to zero if ymin is larger than zero:
if ymin>0 then
       ymin = 0
end;

//Draw plot of function and rectangles, and calculate summation
if stype == 'L' then
       xset('window',1);xbasc(1);
        plot2d1('onn',x',y',[1],'011','y',[xmin ymin xmax ymax]); //[1]
       plot2d2('onn',x',y',[1],'000');                           //[2]
       plot2d3('onn',x',y',[1],'000');                           //[3]
       xtitle('Left sum','x','y');                               //[4]
       I = sum(y(1:m-1))*Dx;                                     //[5]
elseif stype == 'M' then
       xset('window',2);xbasc(2);
       yyy = (y+yy)/2;
       plot2d1('onn',x',y',[1],'011','y',[xmin ymin xmax ymax]);
        plot2d2('onn',x',yyy',[1],'000');
       plot2d3('onn',x',yyy',[1],'000');
       xtitle('Middle sum','x','y');
       I = sum(yyy(1:m-1))*Dx;
else
       xset('window',3);xbasc(3);
       plot2d1('onn',x',y',[1],'011','y',[xmin ymin xmax ymax]);
        plot2d2('onn',x',yy',[1],'000');
       plot2d3('onn',x',yy',[1],'000');
       xtitle('Right sum','x','y');
       I = sum(yy(1:m-1))*Dx;
end;

//     Notes:
//     [1] Plot the curve y=f(x).
//     [2] Plot step function for y = f(x).
//     [3] Plot vertical lines for y = f(x).
//     [4] Plot title and labels.
//     [5] Calculate summation.

//Draw the x-axis if ymin is negative:
if ymin<0 then
```



```
   xpoly([xmin,xmax],[0,0],'lines'); //draw x-axis if any
end;

//end function
```

Applications of function *Sumint* for calculating integrals

We use function *Sumint* to approximate the integral of functions $f(x) = 1-x^2$ and $g(x) = sin(x) + sin(2x)$. First, we define the functions:

```
-->deff('[y]=f(x)','y=1-x^2')
-->deff('[y]=g(x)','y=sin(x)+sin(2*x)')
```

Next, function *Sumint* is loaded into SCILAB: `-->getf('Sumint')`

The following calls to function *Sumint* calculate the integral of f(x) with a = 0 and b = 1.5 using n = 20. The figures illustrating the summation calculations are shown after the function calls.

```
-->Sumint('L',0,1.5,40,f)
 ans  = .4178994
```




```
-->Sumint('M',0,1.5,20,f)
 ans  = .3734418
```




```
-->Sumint('U',0,1.5,20,f)
 ans  = .2846260
```




The actual value of the integral for *f(x)* between *a = 0* and *b = 1.5* is

$$\int_0^{1.5} f(x)dx = \int_0^{1.5} (1-x^2)dx = (x - \frac{1}{3}x^3)\,|_0^1 = 1.5 - \frac{1.5^3}{3}$$

i.e.,

```
-->1.5-1.5^3/3
 ans  =

     .375
```

The values calculated through the summations differ from the actual value of the integral with relative errors of 11.44% for SL_n, 0.41% for SM_n, and 24.09% for SU_n, as shown next:

```
-->error_SLn = (0.4178994-0.375)/0.375*100
 error_SLn  =

    11.43984



-->error_SMn = (0.3734418-0.375)/0.375*100
 error_SMn  =

  -  .41552

-->error_SUn = (0.2846260-0.375)/0.375*100
 error_SUn  =

  - 24.099733
```

The summation SM_n, with an error of less than 1%, provides the best approximation to the integral out of the three summations calculated above. Improved values of the summation approximations can be obtained by increasing the number of sub-intervals in the summation. The following calls to function *Sumint* use values of n = 100 (graphics are omitted):

```
-->Sumint('L',0,1.5,100,f)
 ans  =

     .3919881

-->Sumint('M',0,1.5,100,f)
```



```
 ans  =

     .3749426

-->Sumint('U',0,1.5,100,f)
 ans  =

     .3578972
```

The corresponding errors are:

```
-->error_SL100 = (0.3919881-0.375)/0.375*100
 error_SL100  =

    4.53016

-->error_SM100 = (0.374926-0.375)/0.375*100
 error_SM100  =

  -  .0197333

-->error_SMU100 = (0.357872-0.375)/0.375*100
 error_SMU100  =

  - 4.5674667
```

Thus, SL_n and SU_n show errors of the order of 4.5% for $n = 100$. The middle-sum, SM_n, on the other hand, shows an error of about 2/10%. Thus, the middle sum seems to provide the smallest relative error. Also, as the value of *n* grows larger, the better the approximation to the integral. The following sequence of SCILAB statements use function *Sumint* to approximate the integral with values of n = 10, 100, 1000, 10000, 100000.

```
-->nn = [10,100,1000,10000,100000];

-->SumintSeq = [];

-->for j = 1:5
-->    n = nn(j);
-->    SumintSeq = [SumintSeq Sumint('M',0,1.5,n,f)];
-->end;
```

SCILAB may take longer than usual to finish this operation since the last integral requires calculation 100000 values of y before summing. After the summation calculations end, the sequence of values of the integral is:

```
-->SumintSeq
 SumintSeq  =

!    .3680556     .3749426     .3749994     .3751875     .375 !
```

These results verify the observation that the approximation to the integral improves as the value of n increases.

The following calls to function *Sumint* calculate the integral of $g(x) = sin(x) + sin(2x)$, with $a = 0$ and $b = \pi$ using $n = 40$. The figures illustrating the summation calculations are shown after the function calls.

```
-->Sumint('L',0,%pi,40,g)
 ans  = 1.9989184
```





```
-->Sumint('M',0,%pi,40,g)
 ans  = 1.9989184
```




```
-->Sumint('U',0,%pi,40,g)
 ans  = 1.9989184
```



The actual value of the integral for *g(x)* between *a = 0* and *b =* π is

$$\int_0^{\pi} g(x)dx = \int_0^{\pi} (\sin x + \sin 2x)dx = (-\cos x - \frac{1}{2}\cos 2x)\,|_0^{\pi}$$

i.e.,


```
--> -cos(%pi)-(1/2)*cos(2*%pi)-(-cos(0)-(1/2)*cos(2*0))
ans  =

2.
```

For the function y = g(x) = sin(x) + sin(2x) the three summation approximations to the integral with *n* = *40*, produce results that are very close to the actual value of 2.0.

The examples presented in this section, thus, illustrate the fact that there is always an error involved in estimating an integral through summations. In the next section we present the trapezoidal rule for integration, which reduces the error in the numerical calculation of the integral with respect to the summation of rectangles used in this section.

Trapezoid Rule for Numerical Integration

Numerical methods are often used to calculate integrals, particularly when the integrand does not have a closed-formed anti-derivative. One such integral, which is used very often in probability and statistics, as well as in the study of dispersion of contaminants in fluid flow, is the integral of the Gaussian curve:

$$\int_a^b \exp(-\frac{z^2}{2})dz$$

We will develop herein a general method to estimate the integral,

$$I = \int_a^b f(x)\ dx.$$



As illustrated in the figure above we can divide the interval (a, b), where $a = x_1$, and $b = x_n$, into (n-1) sub-intervals: (x_1,x_2), $(x_2,x_3),\ldots,(x_{n-1}, x_n)$, or (x_i, x_{i+1}) for $i = 1, 2, \ldots, n-1$. Recalling that the integral represents the area under the curve $y = f(x)$ between values of $x = a$ and $x = b$, we can estimate the area under the curve corresponding to sub-interval i as

$$\Delta A_i = 1/2(x_{i+1} - x_i)\,(f(x_i)+f(x_{i+1})) = 1/2 \cdot \Delta x_i \cdot (f(x_i)+f(x_{i+1}))$$

where,

$$\Delta x_i = x_{i+1} - x_i.$$

We can simplify the calculation by taking the interval width to be a constant value, say Δx. In such case, the integral is approximated by

$$I = \sum_{i=1}^{n} \Delta A_i = \frac{1}{2} \cdot [\, f(x_1) + 2 \cdot \sum_{i=1}^{n} f(x_i) + f(x_n)] \cdot \Delta x$$

Because the area calculated in this equation is the area of the trapezoid limited by the sub-interval (x_i, x_{i+1}) , and the curve $y = f(x)$, this method is known as the ***trapezoid rule***.

For the case of a constant sub-interval width the following relationships apply:

$$\Delta x = (x_n - x_1)/(n-1),$$

and

$$x_i = x_1 + (i-1)\Delta x,$$

for $i = 1,2,\ldots,n$.

Trapezoid rule calculation using SCILAB function *inttrap*

SCILAB provides function *inttrap* to calculate the numerical integral of a function using the trapezoid rule. The general call to the function is:

$$[v] = inttrap([x,] y),$$

where x is the vector of increasing x coordinate data, its default value being $x = [1, 2, \ldots, m]$, where m is the size of vector y, y is the vector of y coordinate data, and v is the value of the integral.

The function calculates

$$v = \int_a^b f(x)dx,$$

where f is a function described by the data set of data $y_i = f(x_i)$, with $a = x_1$, and $b = x_n$.

To check the operation of function *inttrap* we will use the functions f(x) and g(x) used in the previous section when approximating integrals through the summation of rectangles. First, we calculate the integral of $f(x) = 1 - x^2$ in the interval $[0,1.5]$ using x increments of $\Delta x = 0.5, 0.25, 0.10, 0.05$, and 0.01, corresponding to $n = 3, 6, 15, 30, 150$ sub-intervals.

```
-->x=[0:0.5:1.5];inttrap(x,f(x))
 ans  =
```



```
     .3125

-->x=[0:0.25:1.5];inttrap(x,f(x))
 ans  =

     .359375

-->x=[0:0.1:1.5];inttrap(x,f(x))
 ans  =

     .3725

-->x=[0:0.05:1.5];inttrap(x,f(x))
 ans  =

     .374375

-->x=[0:0.01:1.5];inttrap(x,f(x))
 ans  =

     .374975
```

The next set of examples shows the integral of function *g(x) = sin(x) + sin(2x)* in the interval *[0, π]*, with values of $\Delta x = \pi/2$, $\pi/3$, $\pi/5$, $\pi/10$, $\pi/20$, and $\pi/100$, corresponding to *n = 2, 3, 5, 10, 20*, and *100*.

```
-->x = [0:%pi/2:%pi];inttrap(x,g(x))
 ans  =

    1.5707963

-->x = [0:%pi/3:%pi];inttrap(x,g(x))
 ans  =

    1.8137994

-->x = [0:%pi/5:%pi];inttrap(x,g(x))
 ans  =

    1.9337656

-->x = [0:%pi/10:%pi];inttrap(x,g(x))
 ans  =

    1.9835235

-->x = [0:%pi/20:%pi];inttrap(x,g(x))
 ans  =

    1.995886

-->x = [0:%pi/50:%pi];inttrap(x,g(x))
 ans  =

    2.0013068

-->x = [0:%pi/100:%pi];inttrap(x,g(x))
 ans  =

    2.0003284
```


As observed in the case of the summation of rectangles presented in the previous section, the approximation of the integral improves as the number of sub-intervals used in the calculation increases.

Additional examples for function *inttrap*

The following examples illustrate different forms for calling function *inttrap.* The exercises are provided for the user to try on his or her own.

In this first example, a call to a SCILAB function, namely *sin(x)*, is included as the second argument to function *inttrap*:

```
x=(0:0.1:1.5);
inttrap(x,sin(x))
```

In the next example, a user-defined SCILAB function is included as the second argument to function *inttrap*:

```
x = (0:0.1:1.0);
deff('[y]=f(x)','y = x^3 - 2*x + sin(x)');
inttrap(x,f(x))
```

In the next three examples, both x and y are defined as vectors before calling function *inttrap*:

```
x = (0.4:0.2:2.0);
y=[5.16, 3.6922, 3.14, 3.0, 3.1067, 3.3886, 3.81,4.3511, 5.0];
inttrap(x,y)

x = (1.0:0.1:2.0);
y=x^(-1) ;
inttrap(x,y)

x=[0., 0.05, 0.1, 0.2, 0.5, 0.6, 0.9, 1.0];
y=[1.2, 1.3, 1.25, 1.45, 2.3, 4.5, 5.0, 1.2];
inttrap(x,y)
```

Plotting the trapezoidal approximation

To illustrate the trapezoidal approximation to an integral we propose the following user-defined function *plottrap* (*plot trap*ezoidal approximation):

```
function plottrap(x,f)

//This function plots the trapezoidal approximation
//to a function y = f(x) as well as plotting the
//original function on the range of values of x.

[n m] = size(x);                    //Size of vector x
y = f(x);                           //Data for trapezoidal fitting
xx = [x(1):(x(m)-x(1))/100:x(m)];   //Generate 100 points xx on x
yy = f(xx);                         //Generate y = f(xs) for xx

xmin = min(xx);                     //Minimum value of xx
xmax = max(xx);                     //Maximum value of xx
ymin = min(yy);                     //Minimum value of yy
```



```
ymax = max(yy);                        //Maximum value of yy

//Change ymin to zero if ymin is larger than zero:
if ymin>0 then
      ymin = 0
end;

//Draw plot of function and trapezoids
xset('window',1);xbasc(1);
plot2d1('onn',xx',yy',[1],'011','y',[xmin ymin xmax ymax]); //[1]
xpoly(x',y','lines');                                      //[2]
plot2d3('onn',x',y',[1],'000');                            //[3]
xtitle('Trapezoid rule','x','y');                          //[4]

//   Notes:
//   [1] Plot the curve y=f(x).
//   [2] Plot trapezoid top for y = f(x).
//   [3] Plot vertical lines for y = f(x).
//   [4] Plot title and labels.

//Draw the x-axis if ymin is negative:
if ymin<0 then
   xpoly([xmin,xmax],[0,0],'lines'); //draw x-axis if any
end;

//end function
```

The call to the function requires that the user provide an ordered vector of values of *x* and the function *f* to be plotted. The general call to the function is *plottrap(x,f)*. An example of its application is shown below:

```
-->xx = [0:%pi/10:%pi];

-->deff('[y]=g(x)','y=sin(x)+sin(2*x)')

-->plottrap(xx,g)
```



Simpson's 1/3 Rule

The summation of rectangles for estimating integrals can be though of as approximating the function to a constant value (the height of the rectangle) in each sub-interval $[x_i,x_{i+1}]$. The trapezoid rule can be thought of as fitting a straight line between the points $(x_i,f(x_i))$ and $(x_{i+1},f(x_{i+1}))$ in each sub-interval. Simpson's 1/3 rule results from fitting a quadratic equation, say,

$$y = px^2 + qx + r,$$

using the three consecutive points $(x_i,f(x_i))$, $(x_{i+1},f(x_{i+1}))$, and $(x_{i+2},f(x_{i+2}))$ that link the two consecutive sub-intervals $[x_i, x_{i+1}]$ and $[x_{i+1}, x_{i+2}]$, as illustrated in the figure below (with $y_i = f(x_i)$):



You can verify that, for sub-intervals of the same length, $\Delta x = x_{i+1} - x_i = x_{i+2} - x_i$, the expression for the integral in the two sub-intervals limited by $[x_i, x_{i+2}]$ is given by

$$\int_{x_i}^{x_{i+2}} f(x)dx \approx \frac{\Delta x}{3}(y_i + 4y_{i+1} + y_{i+2}).$$

A similar expression can be found for the integral in the next pair of sub-intervals $[x_{i+2},x_{i+3}]$ and $[x_{i+3},x_{i+4}]$, i.e.,

$$\int_{x_{i+2}}^{x_{i+4}} f(x)dx \approx \frac{\Delta x}{3}(y_{i+2} + 4y_{i+3} + y_{i+4}).$$

Combining the last two expressions we get an approximation to the integral in the interval $[x_i,x_{i+4}]$, i.e.,

$$\int_{x_i}^{x_{i+4}} f(x)dx \approx \frac{\Delta x}{3}(y_i + 4y_{i+1} + 2y_{i+2} + 4y_{i+3} + y_{i+4}).$$

Following this pattern of summation, we can write the integral in the next 4 sub-intervals as

$$\int_{x_{i+4}}^{x_{i+8}} f(x)dx \approx \frac{\Delta x}{3}(y_{i+4} + 4y_{i+5} + 2y_{i+6} + 4y_{i+7} + y_{i+8}).$$

Adding the last two expressions we can write the integral for the 8 sub-intervals enclosed in $[x_i, x_{i+8}]$ as:

$$\int_{x_i}^{x_{i+8}} f(x)dx \approx \frac{\Delta x}{3}(y_i + 4y_{i+1} + 2y_{i+2} + 4y_{i+3} + 2y_{i+4} + 4y_{i+5} + 2y_{i+6} + 4y_{i+7} + y_{i+8}),$$

or,

$$\int_{x_i}^{x_{i+8}} f(x)dx \approx \frac{\Delta x}{3}[y_i + 4(y_{i+1} + y_{i+3} + y_{i+5} + y_{i+7}) + 2(y_{i+2} + 2y_{i+4} + 2y_{i+6}) + y_{i+8}].$$

We can see the general pattern of the integral in the latter expression. One obvious result is that the number of sub-intervals for the calculation of the integral must be even. This is so that we can fit quadratic equations through pairs of consecutive sub-intervals. Thus, if we divide the interval [*a,b*] into *2n* sub-intervals, referring to the values of *x* limiting those sub-intervals as $x_0(=a)$, x_1, x_2, …, $x_{2n}(=b)$, (make *i=0* in the previous expression) we can write the general expression for Simpson's 1/3 rule as

$$\int_a^b f(x)dx \approx \frac{\Delta x}{3} \cdot [y_0 + 4(y_1 + y_3 + y_5 + \cdots + y_{2n-1}) + 2(y_2 + 2y_4 + 2y_6 + \cdots + y_{2n-2}) + y_{2n}].$$

From the last expression it is obvious why this approach is referred to as the 1/3 rule. The following function, *simpson13*, calculates an integral using Simpson's 1/3 rule given the vectors x and y containing an odd number of elements. Vector x must contain the limits of the even number of sub-intervals into which the integration interval [*a,b*] is divided:

```
function [I] = simpson13(x,f)
//This function calculates the numerical integration of f(x)dx
//between limits x(1) and x(n) using Simpson's 1/3 rule
//Check that x and y have the same size (which must be an odd number)
//Also, the values of x must be equally spaced with spacing Dx
[nrx,ncx]=size(x)
[nrf,ncf]=size(f)
if((nrx<>1)|(nrf<>1))then
       error('x or f, or both, not column vector(s)');
       abort;
end;
if((ncx<>ncf))then
       error('x and f are not of the same length');
       abort;
end;
//check that the size of the vector x and f is odd
if(modulo(ncx,2)==0)then
       disp(ncx,"list size =")
       error('list size must be an odd number');
       abort
end;
n = ncx;
xdiff =  mtlb_diff(x);
h = xdiff(1,1);
I = f(1,1) + f(1,n);
```



```
for j = 2:n-1
      if(modulo(j,2)==0) then
            I = I + 4*f(1,j);
      else
            I = I + 2*f(1,j);
      end;
end;
I = (h/3.0)*I
//end of function simpson13
```

The following commands show applications of the function *simpson13* for calculating integrals. The function *inttrap* is also used to compare the results of the two integration procedures. The main restriction in using the *simpson13* function is that the number of elements in the x,f vectors must be an odd number. Also, the spacing in the values of x must be constant.

```
-->getf('simpson13')              //Load function

-->x = (0:0.1:1.6);               //Generate vector of x values

-->simpson13(x,sin(x))            //Simpson 1/3 rule - Example 1
 ans  =

    1.0292001

-->inttrap(x,sin(x))              //Compare with inttrap
 ans  =

    1.0283417

-->x = (0.4:0.2:2.0);            //New vector of values of x

-->y = [5.16, 3.69, 3.14, 3.0, 3.11, 3.39, 3.81, 4.35, 5.0];

-->simpson13(x,))                //Simpson 1/3 rule - Example 2

 ans  =

    5.8666667

-->inttrap(x,y)                  //Compare with inttrap
 ans  =

    5.914

-->x = (1.0:0.1:2.0);            //New vector of values of x

-->y=x^(-1);                     //Vector y = 1/x

-->simpson13(x,y)                //Simpson 1/3 rule - example 3
 ans  =

     .6931502

-->inttrap(x,y)                  //Compare with inttrap
 ans  =

     .6937714

-->deff('[y]=f(x)','y=x^3-2*exp(x)')  //User-defined function

-->x = (1.0:0.1:2.0);            //New vector of values of x
```



```
-->yp = f(x);                      //Vector of values of y

-->simpson13(x,yp)                 //Simpson 1/3 rule - example 4
 ans  =

  - 5.5915537

-->inttrap(x,yp)                  //Compare with inttrap
 ans  =

  - 5.5918319
```

Simpson's 3/8 Rule

In the previous section we demonstrated that Simpson's 1/3 rule results from fitting quadratic equations to the three points $(x_i, f(x_i))$ defining two consecutive sub-intervals in the integration interval (a,b). Simpson's 3/8 rule follows from fitting cubic equations to the four points that define three consecutive sub-intervals in the integration interval. Skipping the details of the derivation, we present the general form of the integral approximated by Simpson's 3/8 rule as:

$$\int_a^b f(x)dx \approx$$

$$\frac{3}{8}\Delta x \cdot [y_0 + 3(y_1 + y_4 + \cdots + y_{3n-2}) + 3(y_2 + y_5 + \cdots + y_{3n-1}) + 2(y_3 + y_6 + \cdots + y_{3n-3}) + y_{3n}].$$

The method requires that the integration interval be divided into a number of sub-intervals that is a multiple of three. As in the case of Simpson's 1/3 rule, we define $y_i = f(x_i)$, and take $x_0 = a$ and $x_{3n} = b$.

The following function, *simpson38*, calculates an integral using Simpson's 3/8 rule given a vector of equally-spaced values of $x = [a=x_0, x_1, \ldots, x_{3n} = b]$, the corresponding vector of values of $y = [y_0\ y_1\ \ldots\ y_{3n}] = [\ f(x_0)\ f(x_1)\ \ldots\ f(x_{3n})]$. The function checks that the number of sub-intervals is indeed a multiple of 3, producing an error otherwise.

```
function [I] = simpson38(x,f)

//This function calculates the numerical integration of f(x)dx
//between limits x(1) and x(n) using Simpson's 3/8 rule
//Check that x and y have the same size (which must be of the form 3*i+1,
//where i is an integer number)
//Also, the values of x must be equally spaced with spacing h

[nrx,ncx]=size(x);
[nrf,ncf]=size(f);

if((nrx<>1)|(nrf<>1))then
       error('x or f, or both, not column vector(s)');
       abort;
end;
if((ncx<>ncf))then
       error('x and f are not of the same length');
       abort;
end;

//check that the size of the lists xL and f is odd
```



```
if(modulo(ncx-1,3)<>0)then
       disp(ncx,"list size =")
       error('list size must be of the form 3*i+1, where i=integer');
       abort
end;
n = ncx;
xdiff =  mtlb_diff(x);
h = xdiff(1,1);
I = f(1,1) + f(1,n);
for j = 2:n-1
       if(modulo(j-1,3)==0) then
              I = I + 2*f(1,j);
       else
              I = I + 3*f(1,j);
       end;
end;
I = (3/8)*h*I;
//end of function simpson38
```

The following commands show applications of the function *simpson38* for calculating integrals. The function *inttrap* is also used to compare the results of the two integration procedures. The main restriction in using the *simpson38* function is that the number of elements in the x,f vectors must be a number of the form *3*i+1*, where *i* is an integer. Also, the spacing in the values of x must be constant.

```
-->getf('simpson38')          //Load function

-->x=(0:0.1:1.8);             //Vector of values of x

-->simpson38(x,sin(x))        //Simpson's 3/8 rule - example 1
 ans  =

    1.2272036

-->inttrap(x,sin(x))          //Compare with inttrap
 ans  =

    1.2261793

-->x = (0.4:0.2:2.2);         //New vector of values of x

-->y = [5.16, 3.70, 3.4, 3.0, 3.11, 3.39, 3.81, 4.35, 5.0, 5.2];

-->simpson38(x,y)             //Simpson's 3/8 rule - example 2
 ans  =

    6.96225

-->inttrap(x,y)               //Compare with inttrap
 ans  =

    6.988

-->x = (1.0:0.1:1.9);         //New vector of values of x

-->y = x^(-1);                //Vector of values of y = 1/x

-->simpson38(x,y)             //Simpson's 3/8 rule - example 3

 ans  =

     .6418605
```



```
-->inttrap(x,y)               //Compare with inttrap
 ans  =

     .6424556

-->deff('[y]=f(x)','y=x^3-2*exp(x)')  //User defined function for integration

-->x = (1.0:0.1:1.9);         //New vector of values of x

-->yp = f(x);                 //Vector of values of y = f(x)

-->simpson38(x,yp)            //Simpson's 3/8 rule - example 4
 ans  =

  - 4.9272101

-->inttrap(x,yp)              //Compare with inttrap
 ans  =

  - 4.9272868
```

Newton-Cotes Formulas

The trapezoidal rule and Simpson's 1/3 and 3/8 rules are the first three of a collection of formulas based on equally spaced values of the independent variable and commonly known as Newton-Cotes Formulas. The Newton-Cotes formulas are based on fitting polynomials of orders n = 1, 2, 3, etc. to 2, 3, 4, etc. consecutive points determining the sub-intervals into which the integration interval [*a,b*] is divided.. The general expression for the formulas is

$$\int_a^b f(x)dx \approx n \cdot \kappa \cdot \Delta x \cdot (\beta_0 y_0 + \beta_1 y_1 + ...),$$

where *n* represents the number of intervals and the degree of the polynomial fitted, κ and β_i's are coefficients given in the following table:

n	β_0	β_1	β_2	β_3	β_4	β_5	β_6	β_7	κ
1	1	1							1/2
2	1	4	1						1/6
3	1	3	3	1					1/8
4	7	32	12	32	7				1/90
5	19	75	50	50	75	19			1/288
6	41	216	27	272	27	216	41		1/840
7	751	3577	1323	2989	2989	1323	3577	751	1/17280

The following function, *NewtonCotes*, allows the estimation of an integral using the Newton-Cotes formulas up to seventh degree. It is implemented only for functions defined as *y = f(x)* (i.e., the function f must be defined using *deff* or by a file-defined function loaded with *getf*). The function *NewtonCotes* requires the limits of integration (*a,b*), the function name *f(x),* and the degree of the polynomial to be used *n* ($1 \le n \le 7$).


```
function [I] = NewtonCotes(a,b,f,n)

//This function calculates the numerical integration of f(x)dx
//between limits a and b using Newton-Cotes integration of degree n.

if((n<1)|(n>7))then
       disp(n,"n =");
       error('n must be an integer between 1 and 7.');
       abort;
end;

kappa = [1/2, 1/6, 1/8, 1/90, 1/288, 1/840, 1/17280];
beta = [[1 1 0 0 0 0 0 0];
        [1 4 1 0 0 0 0 0];
        [1 3 3 1 0 0 0 0];
        [7 32 12 32 7 0 0 0];
        [19 75 50 50 75 19 0 0];
        [41 216 27 272 27 216 41 0];
        [751 3577 1323 2989 2989 1323 3577 751]];
h = (b-a)/n;
x = zeros(n+1,1);
y = x;
I = 0.0
for j = 1:n+1
       x(j,1) = a + (j-1)*h;
       y(j,1) = f(x(j,1));
       I = I + beta(n,j)*y(j,1);
end;
I = n*kappa(n)*h*I
//end function NewtonCotes
```

The following examples show applications of the Newton-Cotes integration formulas. In each example we define a function using SCILAB's function *deff* and produce a vector containing the integrals using polynomials of orders *n = 1* to *n = 7*:

```
-->getf('NewtonCotes')

-->deff('[y]=f(x)','y=1/x');

-->I = []; for k=1:7, I=[I NewtonCotes(1,2,f,k)]; end; I
 I  =

         column 1 to 6

!    .75     .6944444     .69375     .6931746     .6931630     .6931481 !

         column 7

!    .6931477 !

-->deff('[y]=f(x)','y=x');

-->I = []; for k=1:7, I=[I NewtonCotes(1,2,f,k)]; end; I
 I  =

!   1.5    1.5    1.5    1.5    1.5    1.5    1.5 !

-->deff('[y]=f(x)','y=x^2');

-->I = []; for k=1:7, I=[I NewtonCotes(1,2,f,k)]; end; I
 I  =
```



```
         column 1 to 6

!   2.5    2.3333333    2.3333333    2.3333333    2.3333333    2.3333333 !

         column 7

!   2.3333333 !

-->deff('[y]=f(x)','y=exp(x)');

-->I = []; for k=1:7, I=[I NewtonCotes(1,2,f,k)]; end; I
 I  =

         column 1 to 6

!   5.053669    4.672349    4.6714765    4.6707766    4.6707756    4.6707743 !

         column 7

!   4.6707743 !

-->deff('[y]=f(x)','y=log(x)')

-->I = []; for k=1:7, I=[I NewtonCotes(1,2,f,k)]; end; I
 I  =


         column 1 to 6

!    .3465736     .3858346     .3860838     .3862879     .3862906     .3862942
!

         column 7

!    .3862943 !
```

Romberg Integration

The Romberg algorithm consists in improving the approximation of an integral by first calculating the integral using a constant increment Δx and then re-calculating it with an increment $\Delta x/2$. The two values of the integral thus calculated are then combined through the formula:

$$I(Romberg) = I_{\Delta x/2} + \frac{1}{3}\left[I_{\Delta x/2} + I_{\Delta x}\right],$$

where $I_{\Delta x/2}$ is the integral calculated with a sub-interval size of $\Delta x/2$ and $I_{\Delta x}$ is the integral calculated with a sub-interval size of Δx.

To calculate the values $I_{\Delta x/2}$ and $I_{\Delta x}$ we can simply use the trapezoidal rule (i.e., SCILAB's *inttrap* function), such as illustrated in function *Romberg*, listed below:

```
function [I]=Romberg(a,b,f,h)

//This function calculates the numerical integral of f(x) between
```



```
//x = a and x = b, with intervals h.  Intermediate results are obtained
//by using SCILAB's own inttrap function

x=(a:h:b);
y=f(x);
I1 = inttrap(x,y);
x=(a:h/2:b);
y=f(x);
I2 = inttrap(x,y);
I = I2 + (1.0/3.0)*(I2-I1);
//end function Romberg
```

The call to function *Romberg* is similar to that of *NewtonCotes* except that the increment h is used in place of a polynomial order. Some examples of Romberg integration are shown below:

```
-->getf('Romberg.txt')

-->Dx = [0.1,0.05,0.01,0.005,0.001];

-->deff('[y]=f(x)','y=x^(-1)')

-->I=[];for j=1:5, h=Dx(j); I = [I Romberg(1,2,f,h)]; end; I
 I  =

!    .6931474     .6931472     .6931472     .6931472     .6931472 !

-->deff('[y]=f(x)','y=exp(-0.02*x).*sin(x)')

-->I=[];for j=1:5, h=Dx(j); I = [I Romberg(1,2,f,h)]; end; I
 I  =

!    .9280852     .9280852     .9280852     .9280852     .9280852 !

-->deff('[y]=f(x)','y=x^3-3*x+2')

-->I=[];for j=1:5, h=Dx(j); I = [I Romberg(1,2,f,h)]; end; I
 I  =

!   1.25    1.25    1.25    1.25    1.25 !

-->deff('[y]=f(x)','y=exp(-x^2/2)/sqrt(2*%pi)')

-->I=[];for j=1:5, h=Dx(j); I = [I Romberg(0,1,f,h)]; end; I


 I  =

!    .3413448     .3413447     .3413447     .3413447     .3413447 !

-->deff('[y]=f(x)','y=sin(x)')

-->I=[];for j=1:5, h=Dx(j); I = [I Romberg(0.1,1,f,h)]; end; I
 I  =

!   .4547019    .4547019    .4547019    .4547019    .4547019 !
```


Other integrating functions provided by SCILAB

This section presents functions provided by SCILAB that produce integration by quadrature, by spline fitting, and by definite integration.

Integration by quadrature

By integration by quadrature we must understand the integral of an external function, as opposite to the integral of a function provided as a vector of values, such as those obtained with *inttrap*. Integration by quadrature is accomplished with the function *integrate*. The general call to the function *integrate* is

[x]=integrate(f,v,a,b [,ea [,er]])

where *f* = external SCILAB function, *v* = integration variable entered as a string, *a,b* = real numbers (bounds of integration), and *ea ,er* = real numbers (absolute and relative errors). The presence of errors

Some examples of applications of the integrate function are shown below. The first example is the integral $\int_0^{\pi} sin(x)\,dx$:

```
--> integrate('sin(x)','x',0,%pi)
 ans  =

    2.
```

The following call to *integrate* uses a function defined by an *if…else…end* programming construct:

```
-->integrate('if x<0 then x, else x^2-1, end','x',-2,2)
 ans  =

    4.6666667
```

The following is the integral $\int_0^{\pi} dx/sin(x)$:

```
-->integrate('1/sin(x)','x',%pi/10,%pi/2)
 ans  =1.84273
```

In the next example the function f(x) is defined using *deff*. Notice that the call to function *integrate* includes the name of the function *'f(x)'* as a string.

```
-->deff('[y]=f(x)','y=sin(x)+sin(2*x)')

-->integrate('f(x)','x',0,%pi)
 ans  =  2.
```

The following example is the integral $\int_0^{\pi} dx/x$:

```
-->integrate('1/x','x',1,5)
 ans  =

    1.6094379
```


Integration by spline interpolation

Integration by spline interpolation is accomplished by using the function *intsplin*, whose general call is given by

I = intsplin([x,] y),

where *x* = vector of increasing x coordinate data, y = vector of y coordinate data, and *I* = value of the integral. This function computes the integral of *f(x)* between $x = x_0$ and $x = x_1$, where the values of *f(x)* are represented by the vector *y*. The vector $x = \{x_0, \ldots, x_n\}$, and $y_i = f(x_i)$. The function between discrete values is interpolated using splines. [**Note**: Splines are piece-wise curves used to fit data between consecutive data points. Splines are covered in a different chapter of this book.]

Examples of interpolation using splines follow:

```
-->x = 0:0.1:%pi; y = sin(x) + sin(2*x);

-->intsplin(x,y)
 ans  = 2.0008637

-->x = (-4:0.1:0); y = exp(-x^2/2)/sqrt(2*%pi);

-->intsplin(x,y)
 ans  =  .4999685
```

Calculation of definite integrals

SCILAB offers a function, called *intg*, that calculates the definite integral of a function. The general call to function *intg* is

[I,err]=intg(a,b,f [,ea [,er])

where *a,b* = limits of integration, *f* = represents a function, list or string, *ea* = absolute error required on the result (default value, *ea* = 0), *er* = relative error required on the result (default value, *er* = 1.d-8), *I* = integral value, and *err* = estimated absolute error on the result.

If *f* is function its definition must be as follows: y = f(t). If *f* is a list the list must be as follows: *list(f,x1,x2,...),* where *f* is a function with calling sequence: *f(t,x1,x2,...)*. Finally, if *f* is a string it refers to the name of a Fortran subroutine.

An application of function *intg* follows:

```
-->deff('[y]=f(x)','y=cos(x)+0.5*cos(2*x)+1.5*cos(3*x)')

-->[I,err] = intg(0,1,f)
 err  =
    1.668E-14
 I  =
    1.1393553
```


Integrals of functions of a complex variable

SCILAB provides a couple of functions, *intl* and *intc* for calculation of integrals of functions of a complex variable z = x + iy, where x and y are real variables and $i = \sqrt{-1}$ is the unit imaginary number. While the application of the functions is simple and straightforward, we need to provide some definitions related to complex variables and their functions for the benefit of those who have not covered functions of a complex variable.

Functions of a complex variable

We defined a complex variable z as $z = x + iy$, where x and y are real variables, and $i = (-1)^{1/2}$. We can also define another complex variable

$$w = F(z) = \Phi + i\Psi,$$

where, in general,

$$\Phi = \Phi(x,y), \text{ and } \Psi = \Psi(x,y),$$

are two real functions of (x, y). These real functions can also be given in terms of the polar coordinates (r, θ) if we use the polar representation for z, i.e.,

$$z = r \cdot e^{i\theta} = r\,(\cos\theta + i \cdot \sin\theta).$$

In such case,

$$\Phi = \Phi(r,\theta), \text{ and } \Psi = \Psi(r,\theta).$$

Recall that the coordinate transformations between Cartesian and polar coordinates are:

$$r = (x^2 + y^2)^{1/2}, \qquad \tan\theta = y/x,$$

$$x = r\cos\theta, \qquad y = r \sin\theta$$

The complex variable w is also known as a *complex function*. Another name for a complex function is "*mapping*." Thus, we say $F(z)$ is a mapping of z. In geometric terms, this means that any figure in the x-y plane gets "mapped" onto a different figure on the Φ–Ψ plane by the complex function F(z).

As an example, take the function

$$w = F(z) = \ln(z) = \ln(r \cdot e^{i\theta}) = \ln(r) + i\theta.$$

We can identify the functions

$$\Phi = \Phi(r,\theta) = \ln(r), \text{ and } \Psi = \Psi(r,\theta) = \theta,$$

as the real and imaginary components, respectively, of the function $\ln(z)$. Using the transformations indicated above we can also write,

$$\Phi = \Phi(x,y) = \ln[(x^2+y^2)^{1/2}] = (1/2)\ln(x^2+y^2), \text{ and } \Psi = \Psi(x,y) = \tan^{-1}(y/x).$$

Derivative of a complex function

The derivative of the complex variable *f(z),* to be referred to as *f'(z)* = *df/dz*, is, by definition,

$$f'(z) = \frac{df}{dz} = \lim_{\Delta z \to 0} \frac{f(z+\Delta z) - f(z)}{\Delta z}.$$

The definition of a complex derivative requires us to evaluate the function *f(z)* at a point *P(x,y)* corresponding to *z = x + iy*, and at point *Q(x+Δx, y+Δy),* as illustrated in the figure below.

The figure also illustrates the fact that to get from point z to point z+Δz in the complex x-y plane you can follow a multitude of paths. In general, the value of the derivative will depend on the path we follow to define Δz. Because we want the derivative df/dz to be uniquely defined, we need to find some criteria such that, regardless of the path selected to define z, the value of df/dz remains the same.



In general, we will write Δz = Δx+iΔy. Let's calculate the derivative df/dz utilizing paths for Δz along the x-axis alone, i.e., Δz = Δx, and along the y-axis alone, i.e., Δz = iΔy. Thus, for Δz = Δx, we can write

$$\frac{df}{dz} = \lim_{\Delta x \to 0} \frac{[\Phi(x+\Delta x, y) + i\Psi(x+\Delta x, y)] - [\Phi(x,y) + i\Psi(x,y)]}{\Delta x}$$

$$\frac{df}{dz} = \lim_{\Delta x \to 0} \left(\frac{[\Phi(x+\Delta x, y) - \Phi(x,y)]}{\Delta x} + i\frac{[\Psi(x+\Delta x, y) - \Psi(x,y)]}{\Delta x} \right)$$

You can prove, by expressing the derivative in terms of *Δz = iΔy*, that

$$\frac{df}{dz} = \frac{\partial \Psi}{\partial y} - i\frac{\partial \Phi}{\partial y}.$$

In order for the last two expressions for *df/dz* to be the same, we require that

$$\frac{\partial \Phi}{\partial x} = \frac{\partial \Psi}{\partial y}, \qquad \frac{\partial \Phi}{\partial y} = -\frac{\partial \Psi}{\partial x}.$$

These two equations are known as the *Cauchy-Riemann differentiability cond*itions for complex functions (or, simply, the Cauchy-Riemann conditions). Thus, if the functions $\Phi(x,y) = Re[f(z)]$ and $\Psi(x,y) = Re[f(z)]$, satisfy the Cauchy-Riemann conditions, the derivative $f'(z)=df/dz$ is uniquely defined. In such case, the function f(z) is said to be an *analytical complex function,* and the functions $\Phi(x,y)$ and $\Psi(x,y)$ are said to be *harmonic functions*.

More importantly, if a complex function f(z) is analytical, the rules used for univariate derivatives can be applied to f(z). For example, we indicated earlier that the function

$$w = f(z) = ln\,(z) = ln\,(r \cdot e^{i\theta}) = ln(r) + i\theta.$$

can be written in terms of (x,y) as

$$\Phi = \Phi(x,y) = ln[(x^2+y^2)^{1/2}] = (1/2)\,ln(x^2+y^2),\ \text{and}\ \Psi = \Psi(x,y) = tan^{-1}(y/x).$$

You can check that the functions $\Phi(x,y)$ and $\Psi(x,y)$ satisfy the Cauchy-Riemann conditions. The function $f(z) = ln(z)$ is, therefore, analytical, and its derivative can be calculated by using:

$$\frac{df}{dz} = \frac{d}{dz}(\ln z) = \frac{1}{z}.$$

Note: Most of the functions that we commonly use with real variables, e.g., *exp, ln, sin, cos, tan, asin, acos, atan*, hyperbolic functions, polynomials, *1/x*, square root, etc., are analytical functions when used with the complex variable $z = x + iy$. Thus, the rules of derivatives for these functions are the same as in real variables, e.g., $d(sin(z))/dz = cos(z)$, $d(z^2+z)/dz = 2z+1$, etc.

Integrals of complex functions

If $f(z) = \phi(x,y) + i\,\psi(x,y)$ is a complex analytic function in a given region of the plane, we define the integral of $f(z)$ along a path C from point $z_1 = x_1 + i\,y_1$ to point $z_2 = x_2 + i\,y_2$, as:

$$\int_C f(z)dz = \int_{z_1}^{z_2} f(z)dz = \int_{(x_1,y_1)}^{(x_2,y_2)} [\phi(x,y) + i\psi(x,y)] \cdot (dx + idy)$$

$$= \int_{(x_1,y_1)}^{(x_2,y_2)} (\phi dx - \psi dy) + i\int_{(x_1,y_1)}^{(x_2,y_2)} (\psi dx + \phi dy)$$

where $dz = dx + i\,dy$.

If the integral evaluates to the same value regardless of the nature of path C, we say that the integral is independent of the path. It depends only on the limits of integration z_1 and z_2. We can prove that if the function $f(z)$ is analytic, then the resulting integrals

$$\int_{(x_1,y_1)}^{(x_2,y_2)} (\phi dx - \psi dy) + i \int_{(x_1,y_1)}^{(x_2,y_2)} (\psi dx + \phi dy)$$

are independent of the path. The conditions for independence of the path for these two integrals are

$$\frac{\partial}{\partial y}\phi = -\left(\frac{\partial}{\partial x}\psi\right) \text{ and } \frac{\partial}{\partial x}\psi = \frac{\partial}{\partial y}\phi ,$$

respectively, which are the Cauchy-Riemann conditions necessary for analycity.

Thus, integrals of any analytic function $f(z)$ can be evaluated as a simple univariate integral treating the function z as a real variable and evaluating the complex integration limits, $z_1 = x_1 + i\,y_1$, and $z_2 = x_2 + i\,y_2$, once the anti-derivative of $f(z)$ has been found.

SCILAB functions for integrals of complex functions

SCILAB provides functions *intc* and *intl* for the numerical calculation of integrals of a complex function. Function *intc* integrates the complex function *f(z)* between complex numbers z_1 and z_2 along the straight line connecting the two numbers. Since line integrals of analytic functions are independent of the path, function *intc* can be used to evaluate integrals of any analytic function. The general call of the function is *[y]=intc(z_1,z_2,f)*, where *f* is an external SCILAB function representing the complex function *f(z)*.

The following examples show how to calculate the integral of the complex function *f(z)* between complex numbers z_1 and z_2 along the straight line joining the two points.

```
-->deff('[w]=f(z)','w=1/z');

-->intc(2+3*%i,-5+4*%i,f)
 ans  =

     .5743114 + 1.484058i

-->deff('[w]=f(z)','w=log(z)')

-->intc(2,3,f)
 ans  =

     .9095425

-->intc(2-%i,%i,f)
 ans  =

  -  .7165866 -  .2679858i

-->deff('[w]=f(z)','w=z+2/z^2')

-->intc(2-4*%i,3-%i,f)
 ans  = 9.6 + 5.2i
```


Function *intl* calculates the line integral of the function of a complex variable $w = f(z)$ on the curve *C* represented by the arc of a circle centered at point $z = z_0$ with radius *r* extending between the angles θ_1 and θ_2. The figure below illustrates the geometry of the problem.



The general call to the function is $[y] = intl(\theta_1, \theta_2, z_0, r, f)$, where *f* is the name of a SCILAB external function. The other terms in the function call have been defined above.

The following examples show some cases of line integrals of function of a complex variable on circular arc paths.

```
-->deff('[w]=f(z)','w=z^2')
-->intl(%pi/6,%pi/2,2-3*%i,0.5,f)
 ans  =

    22.82069 - 35.377982i

-->intl(0.1,3.14,0,1.2,f)
 ans  =

  - 1.1262672 -  .1674675i

-->deff('[w]=f(z)','w=1/z+z')

-->intl(%pi/6,%pi/6+2*%pi,2+%i,1,f)
 ans  =

  - 25.132741 + 25.132741i
```

The current definition of function f(z) has a discontinuity a z = 0, so if we select the center of the circle as $z_0 = 0$, the integral does not converge:

```
-->intl(0,2*%pi,0,0.5,f)
 !--error    24
convergence problem...
at line      19 of function intl                    called by :
intl(0,2*%pi,0,0.5,f)
```

Other examples of function *intl* follow:

```
-->intl(0.1,2*%pi+0.1,2+2*%i,0.5,f)
 ans  = - 50.265482 + 6.2831853i

-->intl(0.1,2*%pi+0.1,0.1+0.1*%i,0.5,f)
 ans  =
```



```
  -  .1256637 + 6.2831853i
```

The following example of an integral path centered at $z_0 = 0.2i$ with radius r = 0.5 fails to converge:

```
-->intl(0.1,2*%pi+0.1,0.2*%i,0.5,f)
 !--error    24
convergence problem...
at line      19 of function intl                    called by :
intl(0.1,2*%pi+0.1,0.2*%i,0.5,f)
```

However, moving the center of the circle to $z_0 = 0.1(1+i)$ produces a convergent integral:

```
-->intl(0.1,2*%pi+0.1,0.1+0.1*%i,0.5,f)
 ans  =

  -.1256637 + 6.2831853i
```

Applications of integrals of one variable

In this section we present applications of integrals for the solution of problems from the physical sciences.

Areas under curves

A simple physical interpretation of the integral $I = \int_a^b f(x)\,dx$, is the area under the curve $y = f(x)$ and above the x-axes, between the values $x = a$ and $x = b$.



A representative differential of area (or area element) is the rectangle of base dx and height $y = f(x)$ shown in the figure. We will write the area of the differential element as $dA = f(x)\,dx$. The total area is,

$$A = \int_R dA = \int_a^b f(x)dx$$

For example, to find the area between x = 2 and x = 4, under the curve $y = ln(x+1)$, use:


```
-->integrate('log(x+1)','x',2,4)
 ans  =   2.7513527
```

The region under consideration can be depicted as shown below:

```
-->x=[0:0.1:6];y=log(x+1);
-->xvv = [2:0.1:4];yvv=log(xvv+1);
-->xva=[2 2 xvv 4 4];yva=[0 log(3) yvv log(5) 0];
-->plot(x,y,'x','y','Integral example')
-->xfpoly(xva,yva,2)
```



Area between curves

The area between the curves $y = f_1(x)$ and $y = f_2(x)$, and $x = a$ and $x = b$, assuming that in (a,b), $f_2(x) \le f_1(x)$, is given by $\int_a^b f_1(x) - f_2(x)\, dx$.

For example, let $y = f_1(x) = x$ and $y = f_2(x) = x^2$, for $a=0.25$ and $b=0.75$. The following statements define the functions in SCILAB and produce the graph of the area to be calculated:

```
-->deff('[y]=f1(x)','y=x');deff('[y]=f2(x)','y=x^2');a=0.25;b=0.75;
-->x=[a:0.01:b];y1=f1(x);y2=f2(x);
-->xv1=[a,a];yv1=[f1(a),f2(a)];xv2=[b,b];yv2=[f1(b),f2(b)];
-->plot2d([x' x'],[y1' y2'],[1,2],'011',' ',[0 0 1 1])
-->xsegs(xv1,yv1,3);xsegs(xv2,yv2,4);
-->xtitle('Area between y1=x and y2 = x^2, 0.25<x<0.75','x','y')
```




The differential of area used for this exercise is shown below.



To calculate the area we use SCILAB function *inttrap*:

```
--> x = [0.25:0.01:0.75];y1=f1(x);y2=f2(x);y=y1-y2;inttrap(x,y);

ans  =  0.11475
```

Center of mass of an area

Consider the rectangular element of area in the figure below, *dA = f(x) dx*. If the base of the rectangle is centered at x, then the center of mass of the rectangle is the point

$$C[x, \frac{f(x)}{2}] ,$$

i.e.,

$$x_c = x \text{ , and } y_c = \frac{\mathrm{f}(x)}{2} \quad .$$

First moments of the areas with respect to the x-and y-axes are:

$$M_x = \int y_c \, dA \; = \; \int_a^b \frac{\mathrm{f}(x)\,\mathrm{f}(x)}{2}\, dx = \; \frac{1}{2} \; \int_a^b \mathrm{f}(x)^2 \, dx \text{ , and } M_y = \int x_c \, dA \; = \; \int_a^b x\,\mathrm{f}(x)\, dx$$

.



For example, for the region *0 < y < ln(1+x), 2<x<4*:

```
-->deff('[y]=f(x)','y=log(x+1)');a=2;b=4;

-->Mx = integrate('f(x)^2','x',a,b)/2

 Mx  = 1.9139498

-->My = integrate('x*f(x)','x',a,b)

 My  =  8.4228659
```

The center of mass has coordinates,

$$X_c = \frac{M_y}{A} \text{ , and } Y_c = \frac{M_x}{A} \quad .$$

Where,

$$A = \int_a^b \mathrm{f}(x)\, dx$$

```
-->A = integrate('f(x)','x',a,b)
 A  =  2.7513527

-->xc = My/A, yc = Mx/A
 xc  = 3.0613545
 yc  =  .6956396
```


The following plot shows the center of mass (centroid) of the area

```
-->x=[2:0.1:4];y=log(x+1);xv1=[2 2];yv1=[0 log(3)];xv2=[4 4];yv2=[0 log(5)];
-->plot2d(x,y,1,'011',' ',[0 0 6 2]);xsegs(xv1,yv1,1);xsegs(xv2,yv2,1);
-->xset('mark',-1,1);plot2d([xc],[yc],-1,'010',' ',[0 0 6 2])
-->xstring(xc,yc,'C')
-->xtitle('Centroid of an area','x','y')
```



Volume of a solid of revolution

The figure below shows a body of revolution generated by the rotation of the curve *y = f(x), a<x<b*, about the x-axis. The shadowed rectangle represents the area differential of width dx, and height f(x), centered at x. The rotation of that differential of area about the x-axis generates a differential of volume in the shape of a disk (a cylinder) of height dx and base area equal to $\pi\, \mathrm{f}(x)^2$. The volume of the elementary disk is $dV = \pi\, \mathrm{f}(x)^2\, dx$. Therefore, the volume of the body of revolution is $V = \int 1\, dV = \int_a^b \pi\, \mathrm{f}(x)^2\, dx$



or example, for the region used earlier to find the centroid of the area, the volume of a body of revolution generated by the rotation of the curve $y = ln(1+x)$, 2<x<4, about the x axis is given by:

```
-->V = integrate('%pi*f(x)^2','x',a,b)

V  =  12.025702
```

Moment of inertia of an area

Consider the rectangular region defined by 0<x<B, 0<y<H, as shown below:



The horizontal element of area shown in the left-hand side figure, is a rectangle of width dy and height L, and centered at y. The element's area is $dA = B\,dy$. The distance from the area element to the x-axis is y, therefore, the moment of inertia of the area element with respect to the x-axis is $dI_x = y^2\,dA = y^2\,B\,dy$. The moment of inertia of the area with respect to the x-axis is

$$I_x = \int 1\,dI_x = \int y^2\,dA = \int_0^H y^2\,B\,dy\,,$$

i.e.,

$$I_x = \frac{1}{3} B\,H^3$$

For the vertical element shown in the right-hand side figure above, we can write $dA = H\,dx$. Its moment of inertia with respect to the y-axis is, $dI_y = x^2\,dA = x^2\,H\,dx$, and the total moment of inertia with respect to the y-axis is

$$I_y = \int 1\,dI_y = \int x^2\,dA = \int_0^L x^2\,H\,dx\,,$$

i.e.,

$$I_y = \frac{1}{3} H B^3$$

These two results can be used to obtain moments of inertia of areas limited by curves in the x-y plane. For example, for the element of area shown in the figure below we can write,

$$B = dx,$$

$$H = f(x),$$

$$dIx = \frac{1}{3} \quad dx \quad \mathrm{f}(x)^3 \ ,$$

and

$$I_x = \frac{1}{3} \int_a^b \mathrm{f}(x)^3 \, dx \ .$$



For example, to find the moment of inertia of the semi-circle shown below with respect to the x-axis, we use:




```
-->deff('[w]=f(x)','w=sqrt(25-x^2)')
-->Ix = integrate('f(x)^3','x',-5,5)/3
 Ix  = 245.43693
```


In general, for a semi-circle of radius R:

$$\frac{1}{3}\int_{-R}^{R} f(x)^3\,dx$$

$$Inertia_x := \frac{1}{8} R\sim^4 \pi$$

Suppose that we want to find the moment of inertia of the same semicircle, but with respect to the x-axis, we will use the element of area shown in the figure below:



The element of area actually consists of two rectangles, of base dy, height $x_1(y)$ and $x_2(y)$, centered at y. The values of the functions $x_1(y)$ and $x_2(y)$ are obtained by solving for x in the equation, $x^2 + y^2 = 25$, thus, $x_1 = \sqrt{25-y^2}$, $x_2 = -\sqrt{25-y^2}$.

The moment of inertia of the right-hand side element, is $dI_{y,1} = \frac{1}{3}\ dy\ |x_1(y)|^3$, while that of the left-hand side element is, $dI_{y,2} = \frac{1}{3}\ dy\ |x_2(y)|^3$. The absolute value is used to indicate that the moment of inertia of the element is based on that of a rectangle, where only the rectangle's dimensions, and not their signs are important. Since, $|x_1(y)| = \sqrt{25-y^2} = |x_2(y)|$, then the moment of inertia of the two elements is $dI_y = dI_{y,1} + dI_{y,2} = \frac{2}{3}\ (25-y^2)^{\left(\frac{3}{2}\right)} dy$, and the moment of inertia of the entire semicircle, with respect to the y axis, is

$$I_y = \frac{2}{3}\int_0^5 (25-y^2)^{\left(\frac{3}{2}\right)} dy$$, i.e.,

```
-->Iy=(2/3)*integrate('(25-y^2)^(3/2)','y',0,5)
 Iy  = 245.43693
```

This result turns out to be the same as the moment of inertia of the semicircle with respect to the x-axis. Since the moment of inertia is a measure of the resistance of a body (or an area) to angular acceleration about the axis of interest, in the same sense that a mass is a measure of the resistance of a body to linear acceleration, the fact that $I_x = I_y$ indicates that is equally

difficult to provide the same angular acceleration to a thin body shaped like a semicircle whether you try to make it rotate about the x- or the y-axis.

Parallel axes theorem

The figure below shows an area in the x-y plane indicating its centroid and a couple of axes, x_c, y_c , parallel to the x- and y-axes, respectively, and passing through the area's centroid. The distance between the x- and x_c -axes is d_y , and that between the y- and y_c -axes is d_x .



The parallel axes theorem states that if I_x represents the moment of inertia of the area with respect to the x-axis, and I_{xc} represents the moment of inertia of the area with respect to the x_c -axis, the following relationship holds: $I_x = I_{xc} + A\, d_y^{\,2}$
Similarly, for the moment of inertia with respect to the y-axis, I_y , in terms of the moment of inertia with respect to the y_c -axis, we can write, $I_y = I_{yc} + A\, d_x^{\,2}$

Centroidal moments of inertia for a rectangle

Earlier we found the moments of inertia of a rectangle whose sides lay on the coordinate axes, width (along the x-axis) = L, height (along the y-axis) = H, with respect to those axes as, $I_x = \frac{1\, B\, H^3}{3}$, and $I_y = \frac{1\, H\, B^3}{3}$. With, $A = H\, L$, and $d_x = \frac{B}{2}$, and $d_y = \frac{H}{2}$, we can find the centroidal moments of inertia of a rectangle as shown below:

The parallel axes theorem indicates that: $I_x = I_{xc} + Ad_y^2$, $I_y = I_{yc} + Ad_x^2$.

For the rectangle:

$$Inertia_x := \frac{1}{3} B\, H^3$$

$$Inertia_y := \frac{1}{3} H B^3$$

$$A := B H$$

$$d_x := \frac{1}{2} B$$

$$d_y := \frac{1}{2} H$$

Solving for the centroidal moments of inertia:

$$\{ Inertia_{xc} = \frac{1}{12} B H^3 \}$$

$$\{ Inertia_{yc} = \frac{1}{12} H B^3 \}$$

Moment of inertia of the semicircle using centroidal axes

Noticing that moment of inertia of the rectangle of height can attack the problem of finding the moment of inertia of the semicircle used earlier with respect to the y-axis $|x_1(y)|+|x_2(y)|$, and width dy, about the y-axis is the centroidal moment of inertia, i.e.,

$$dI_y = dI_{yc} = \frac{1}{12} \; dy \; (|x_1(y)|+|x_2(y)|)^3 = \frac{1}{12} \; (2\sqrt{25-y^2})^3 = \frac{2}{3} \; (25-y^2)^{\left(\frac{3}{2}\right)} dy \, ,$$

which is the same value for dI_y found earlier.

Area enclosed by a curve in polar coordinates

Consider the area enclosed by the curve r = r(θ) between the rays $\theta = a$, and $\theta = b$. An element of area in polar coordinates is shown in the figure below. The area of the element shown is approximated by that of a triangle of height r(θ) and base equal to $(r\, d) . \theta$, i.e.,

$$dA = \frac{1}{2} \; (\mathrm{r}(\theta) \; d \, . \, \theta) \, \mathrm{r}(\theta) = \frac{1}{2} \; \mathrm{r}(\theta)^2 \, \mathrm{d}\, \theta \, .$$



For example, find the area enclosed by the curve *r(θ) = 5 cos(θ)*, between the values of $\theta = \frac{\pi}{10}$ and $\frac{\pi}{2}$, we use the integral

```
-->th1 = %pi/10; th2 = %pi/2; A = integrate('0.5*(5*cos(th))^2','th',th1,th2)
 A  = 6.0171527
```

Fluid dynamics: calculating discharge in pipe for laminar flow

The figure below shows the profile of laminar flow velocity as a function of the radial distance r in a pipe.



The velocity distribution is given by the expression,

$$v(r) = v_s.[1-(r/r_o)^2],$$

where v_c is the centerline velocity and r_o is the radius of the pipe.

We can use this expression to obtain the discharge (volumetric flow) in the pipe by using the definition

$$Q = \int_A v \cdot dA.$$

Because the velocity distribution in a pipe depends on the radial distance only, we can use an element of area consisting of a ring of thickness *dr* and length $2\pi r$, thus, the area is

$$dA = 2\,\pi\, r dr.$$

With this element of area, the discharge is calculated, in general, as

$$\int_0^R \mathrm{v}(r)\, 2\,\pi\, r\, dr$$

For the specific case of a laminar flow velocity distribution, you will need to set up the integral:

$$\int_0^R vc\left(1-\left(\frac{r}{R}\right)^2\right) 2\,\pi\, r\, dr$$

The result is $\frac{vc\, r0^2\,\pi}{2}$. The mean velocity is defined as V = Q/A, with A = $\pi\, r0^2$, then V = $v_c/2$ for laminar flow.

Line integrals

A line integral is an integral calculated along a curve C in space or in the plane with the integrand defined at each point of C. Suppose that the curve C has the representation

$$\boldsymbol{r}(s) = x(s)\,\boldsymbol{i} + y(s)\,\boldsymbol{j} + z(s)\,\boldsymbol{k},\ a \le s, s \le b\ ,$$

so that $\boldsymbol{r}(s)$ is continuous and has a non-zero, continuous first derivative $\mathrm{d}\boldsymbol{r}/ds$ for $a \le s, s \le b$. Then C is said to be a *smooth curve*, i.e., C has a unique tangent at each of its points whose direction varies continuously as we travel along the curve. Let $\mathrm{f}(x, y, z)$ be a function defined (at least) at each point of C, and let f be a continuous function of s (i.e., $f[x(s),y(s),z(s)] = g(s)$). Then, the *line integral* of f along C from A to B (where $s = a$, and $s = b$, respectively) is calculated as

$$\int_C f(x,y,z)ds = \int_a^b f[x(s),y(s),z(s)]ds$$

The curve C is called the *path of integration*.

For a line integral over a closed path C, the symbol $\oint_C$ is sometimes used in the literature.

In the exercises presented in next sub-section is it assumed that all the integration paths are **piecewise smooth**, i.e., it consists of finitely many smooth curves.

Some properties of line integrals are shown below.

$$\int_C kf\,ds = k\int_C f\,ds;\ \int_C (f+g)ds = \int_C f\,ds + \int_C g ds;\ \int_C f\,ds = \int_{C_1} f\,ds + \int_{C_2} f\,ds;$$

where C is composed of C_1 and C_2.

Evaluation of line integrals

The formula for evaluating a line integral over a curve C when the curve is described by the vector **r**(s), was given earlier as:

$$\int_C f(x,y,z)ds = \int_a^b f[x(s),y(s),z(s)]ds$$

If the curve is given by a parameter *t*, so that the curve C is defined by **r** = **r**(t), with $t_0 \le t, t \le t_1$, then it can be shown that

$$\int_C f(x,y,z)ds = \int_{t_0}^{t_1} f[x(t),y(t),z(t)]\left(\frac{ds}{dt}\right)dt$$

Example 1

Integrate the function $f(x,y) = 3(x^2+y^2)$, along the path y = x from (0,0) to (1,1).
First, we sketch the path of integration:



For the path of integration $y = x$, we can arbitrarily set $x(s) = s$, and $y(s) = s$, and use

$$\int_C f(x,y)ds = \int_a^b f[x(s),y(s)]ds$$

with *a = 0, b = 1*, i.e., $f[x(s),y(s)] = g(s) = 3(s^2+s^2) = 6s^2$, and the integral reduces to

$$\int_C f(x,y)ds = \int_0^1 6s^2 ds.$$

Using SCILAB:

```
--> integrate('6*s^2','s',0,1)

ans = 2
```


Example 2

Calculate the line integral of $f(x,y) = k(x^2 + y^2)$, *k = 1,* counterclokwise along the circle $x^2 + y^2 = r^2$, with *r*=5, from (*0,r*) to (*r,0*): One possible representation of the circle $x^2 + y^2 = r^2$ is the parametric equations, $x = r\cos(s)$, and $y = r\sin(s)$, with $s = \frac{\pi}{2}$ at (*0,r*) and $s = 0$ at (*r,0*). Thus, $f[x(s),y(s)] = k(r^2 \cdot \sin^2 s + r^2 \cdot \cos^2 s) = kr^2$. The line integral is simply,

$$\int_C f(x,y)ds = \int_{\pi/2}^{0} kr^2 ds = -\frac{1}{2}\pi kr^2.$$

Example 3

Using the curve of *Example 2* that the with the integrand being $f(x,y) = x^2\sin(y)$, we replace $x = r\cos(s)$, and $y = r\sin(s)$, in the function *f(x,y)* to obtain $f[x(s),y(s)] = r^2\cos^2 s \sin(r\sin(s))$, with *s* from *a* = $\pi/2$ to *b* = *0*. The integral, using SCILAB, can be calculated as follows:

```
--> integrate('r^2*cos(s)^2*sin(r*sin(s))','s',%pi/2,0)
ans = -6.3445402
```

Example 4

Evaluate the line integral of $f(x,y) = xy^3$, along the line represented by $\boldsymbol{r(t)} = t\,\boldsymbol{i} + 2t\,\boldsymbol{j}$, for $-1 \le t, t \le 1$.

The formula to use is

$$\int_C f(x,y,z)ds = \int_{t_0}^{t_1} f[x(t),y(t),z(t)]\left(\frac{ds}{dt}\right)dt$$

Notice that $\frac{ds}{dt} = \left|\frac{dr}{dt}\right| = |v| = v_s$, the speed of a particle moving with position vector **r**(t) if t represents time. The velocity vector is

$$\boldsymbol{v} = d\boldsymbol{r}/dt = \boldsymbol{i} + 2\boldsymbol{j},$$

and the speed is

$$v_s = ds/dt = |d\boldsymbol{r}/dt| = \sqrt{5}.$$

We also need to define *f[x(t),y(t)],* with *x(t) = t*, and *y(t) = 2t*, i.e., $f[x(t),y(t)] = (t)(2t)^3 = 8t^4$, thus, the integral reduces to

$$\int_C f(x,y)ds = \int_{t=-1}^{t=1} 8t^4 \cdot \sqrt{5}dt.$$

Using SCILAB:

```
-->integrate('8*t^4*sqrt(5)','t',-1,1);
```

ans = 7.1554175

==========

In many applications the integrands of the line integrals are of the form $g(x,y,z)\ \frac{dx}{ds}$, $g(x,y,z)\ \frac{dy}{ds}$, or $g(x,y,z)\ \frac{dz}{ds}$, where s = variable expressing a length on the integration path, and x=x(s), y=y(s), z=z(s). Then, the line integrals are

$$\int_C g(x,y,z)\frac{dx}{ds}ds = \int_C g(x,y,z)ds,$$

and similarly for the other two cases.

For sums of these types of integrals along the same path C we use the following notation:

$$\int_C f\,dx + \int_C g\,dx + \int_C h\,dx = \int_C (f\,dx + g\,dy + h\,dz)$$

==========

Example 5

Evaluate the line integral whose integrand is $[\,x^2 y\,dx + (x-z)\,dy + x\,y\,z\,dz\,]$ where C is the arc of the parabola $y = x^2$ in the plane z = 2 from A[0,0,2] to B[1,1,2].
Here is a plot of the path of integration:



We can calculate the integral by writing all the terms in terms of x, i.e., replacing $y = x^2$, $dy = 2\,x\,dx$, z = 2, $dz = 0$. Therefore,

$[\,x^2 y\,dx + (x-z)\,dy + x\,y\,z\,dz\,] = [\,x^2 x^2\,dx + (x-2)\,2\,x\,dx + x\,y\,z\,0\,] = (x^4 + 2\,x\,(x-2))\,dx$.

The integral is now,

$$\int_0^1 (x^4 + 2x(x-2))dx.$$

To calculate the integral using SCILAB:

```
-->integrate('x^4+2*x*(x-2)','x',0,1)
ans  = -1.1333333
```


Line integrals independent of path

Definitions:

(1) An expression of the form $f\,dx + g\,dy + h\,dz$ is known as a first-order differential form.

(2) If $f = \frac{\partial}{\partial x}u,\ g = \frac{\partial}{\partial y}u,\ h = \frac{\partial}{\partial z}u$, then the differential form shown above is said to be an **exact differential** of a function $u(x, y, z)$, i.e., $du = f\,dx + g\,dy + h\,dz$, or

$$du = \left(\frac{\partial}{\partial x}u\right)dx + \left(\frac{\partial}{\partial y}u\right)dy + \left(\frac{\partial}{\partial z}u\right)dz\ .$$

In general, a line integral of a differential form, i.e., $\int f\,dx + g\,dy + h\,dz$, over a path C will be dependent not only on the endpoints of the integration (say from point P to point Q), but also on the path followed to perform that integration, i.e., on the curve C. However, if the integrand is an exact differential, then the line integral is independent of the path and we can write:

$$\int_C (f\,dx + g\,dy + h\,dz) = \int_P^Q du = u\Big|_Q^P = u(P) - u(Q)$$

For example, the expression

$$\mathbf{e}^x \cos(y)\,dx - \mathbf{e}^x \sin(y)\,dy + 2\,dz$$

is an exact differential form, which you can prove by checking the conditions

$$\frac{\partial}{\partial y}h = \frac{\partial}{\partial z}g\ , \qquad \frac{\partial}{\partial z}f = \frac{\partial}{\partial x}h\ , \qquad \frac{\partial}{\partial x}g = \frac{\partial}{\partial y}f\ .$$

Therefore, $\mathbf{e}^x \cos(y)\,dx - \mathbf{e}^x \sin(y)\,dy + 2\,dz = du$, is an exact differential. To find the function $u(x, y, z)$ we write: $\frac{\partial}{\partial x}u = f = \mathbf{e}^x \cos(y)$, and integrate with respect to x, i.e.,

$$u := \mathbf{e}^x \cos(y) + C(y, z)$$

Next, take the derivative of this function with respect to y and make it equal to g(x,y), i.e.,

$$-\mathbf{e}^x \sin(y) + \left(\frac{\partial}{\partial y}C(y, z)\right) = -\mathbf{e}^x \sin(y)$$

which indicates that,

$$\frac{\partial}{\partial y}C(y, z) = 0$$

Integrating with respect to y, we get:

$$C(y,z) = E(z) + K$$

Thus,

$$u := \mathbf{e}^x \cos(y) + E(z) + K$$

No
w, take the derivative of this last expression with respect to z and make it equal to *h*, i.e.,

$$\frac{\partial}{\partial z} E(z) = 2$$

Which indicates that

$$E(z) = 2z$$

Therefore, the function $u(x,y,z)$ is:

$$u := \mathbf{e}^x \cos(y) + 2z + K$$

If we want to evaluate the line integral of $\mathbf{e}^x \cos(y)\,dx - \mathbf{e}^x \sin(y)\,dy + 2\,dz = du$ between points P(0,0,0) and Q($2, \pi, 2$), we only need to evaluate u(x,y,z) at Q and P and take the difference, i.e.,

```
--> exp(2)*cos(%pi)+ 2*2 - exp(0)*cos(0) + 2*0
ans  =   -4.3890561
```

Work of a force as a line integral

A force field can be represented by a vector function ***F****(x,y,z) = f(x,y,z)****i*** + *g(x,y,z)****j*** + *h(x,y,z)****k***. If the force acts on a particle of mass *m* along a path defined by a curve C starting at point P and ending at point Q, the work exerted by the force field on the particle is, by definition,

$$W = \int_C \mathbf{F} \bullet d\mathbf{r} = \int_P^Q [\, f(x,y,z)dx + g(x,y,z)dy + h(x,y,z)dz],$$

where

$$d\mathbf{r} = dx\,\mathbf{i} + dy\,\mathbf{j} + dz\,\mathbf{k}.$$

If the quantity ***F****•****dr*** is an exact differential, there exists a function *U(x,y,z)* such that

$$dU = \mathbf{F} \bullet d\mathbf{r} = f(x,y,z)dx + g(x,y,z)dy + h(x,y,z)dz,$$

thus, the integral for calculating the work exerted by force ***F****(x,y,z)* between points *P* and *Q* is independent of the path. The function *U(x,y,z)* represents the *potential energy function* associated with the force ***F****(x,y,z)*, and the force field itself is said to be *conservative*. The force field can be obtained out of the potential energy function by using the gradient operator, i.e.,

$$\mathbf{F}(x,y,z) = \frac{\partial U}{\partial x}\mathbf{i} + \frac{\partial U}{\partial x}\mathbf{j} + \frac{\partial U}{\partial x}\mathbf{k}.$$

For example, for a potential energy function $U(x,y,z) = x^2+y^2+z^2$, the force field is given by $\mathbf{F}(x,y,z) = 2(x\mathbf{i}+y\mathbf{j}+z\mathbf{k})$. To calculate the work performed by the force $\mathbf{F}(x,y,z)$ along a path given by $x(s) = 2 \sin s$, $y(s) = 2 \cos s$, and $z = 2s$, from point $P(0,2,0)$, for $s=0$, to point $Q(\sqrt{3},1, 2\pi/3)$, for $s= \pi/3$, we calculate the integrand

$$\mathbf{F}\bullet d\mathbf{r} = 2xdx + 2ydy + 2zdz = 2(2 \sin s)(2 \cos s\, ds) + 2(2 \cos s)(-\ 2 \sin s\, ds) + 2\,(2s)\,(2\, ds),$$

$$\mathbf{F}\bullet d\mathbf{r} = (8 \sin s \cos s - 8 \sin s \cos s + 8s)ds = 8s\, ds.$$

The work performed by the force is

$$W = \int_{s=0}^{s=\pi/3} F \bullet dr = \int_0^{\pi/3} 8sds = 4s^2\,|_0^{\pi/3} = \frac{4\pi^2}{9}.$$

Alternatively, since the force field is conservative,

$$W = U(Q) - U(P) = 3+1+4\pi^2/9 - 4 = 4\pi^2/9.$$

Double integral in a rectangular domain

As the integral of a single variable, say, $\int_a^b f(x)dx$, can be interpreted as the area under the curve y = f(x) between x = a and x = b, the double integral

$$I = \int_a^b \int_c^d f(x,y)dydx,$$

can be interpreted as the volume of the solid that extends between the x-y plane and the surface defined by $z = f(x,y)$ above the rectangular region $R = \{ a < x < b,\ c < y < d \}$, as illustrated in the figure below.



To approximate the volume representing the double integral we first divide the region R into a grid of x and y values as shown in the figure below. There will be *n* sub-intervals in x and *m* in y. Consider the volume element limited by $x_i < x < x_{i+1}$ and $y_j < y < y_{j+1}$. The base of the volume element is the shaded rectangle shown in the figure.

The height of the element, h_{ij}, is taken to be the average of the four values of the function $z_{ij} = f(x_i,y_j)$ evaluated in the four corners of the element of area, i.e.,

$$h_{ij} = \frac{1}{4}\left(z_{ij} + z_{i,j+1} + z_{i+1,j} + z_{i+1,j+1}\right),$$

for *i = 0, 1, 2, …, n-1*, and *j = 0, 1, 2, …, m-1*. Thus, the element of volume corresponding to the shaded rectangle in the x-y plane is

$$\Delta V_{ij} = h_{ij} \cdot \Delta x_i \cdot \Delta y_j = \frac{1}{4}\left(z_{ij} + z_{i,j+1} + z_{i+1,j} + z_{i+1,j+1}\right) \cdot \Delta x_i \cdot \Delta y_j,$$

where $\Delta x_i = x_{i+1} - x_i$, and $\Delta y_j = y_{j+1} - y_j$. In practice, the values of Δx_i and Δy_j are typically taken as constant values although not necessarily the same. Thus, the double integral is approximated by

$$V \approx \sum_{i=1}^{n}\sum_{j=1}^{m} V_{ij} = \frac{1}{4}\Delta x \cdot \Delta y \cdot \sum_{i=1}^{n}\sum_{i=1}^{m}\left(z_{ij} + z_{i,j+1} + z_{i+1,j} + z_{i+1,j+1}\right).$$



User-defined function for calculating double integrals

The function *DoubleIntegral*, shown below, calculates a double integral over a rectangular domain using the approximation outlined above. The function call requires the *x* integration limits and the number of *x* sub-intervals, *a, b, n*, the y integration limits and number of *y* sub-intervals, *c, d, m*, and the function name, *f*. The function *f* is a SCILAB external function (e.g., a function defined with command *deff.*)


```
function [I]=DoubleIntegral(a,b,n,c,d,m,f)

//This function calculates the double integral of f(x,y)
//in the rectangular domain a < x < b, c < y < d through a
//generalization of the trapezoidal rule.

Dx = (b - a)/n;
Dy = (d - c)/m;
x = zeros(1,n+1);
y = zeros(1,m+1);
F = zeros(n+1,m+1);

for i = 1:n+1
      x(1,i) = a + (i-1)*Dx;
end;

for j = 1:m+1
      y(1,j) = c + (j-1)*Dy;
end;


for i = 1:n+1
      for j = 1:m+1
            F(i,j) = f(x(1,i),y(1,j));
      end;
end;
I = F(1,1) + F(1,m+1) + F(n+1,1) + F(n+1,m+1);
for i = 2:n
      I = I + 2*(F(i,1) + F(i,m+1));
end;
for j = 2:m
      I = I + 2*(F(1,j) + F(n+1,j));
end;
for i = 2:n
      for j = 2:m
            I = I + 4*F(i,j);
      end;
end;
I = I*Dx*Dy/4;
//end of DoubleIntegral function
```

Applications of function *DoubleIntegral*

The following commands show applications of the function *DoubleIntegral*. First, we load the function

```
-->getf('DoubleIntegral.txt')
```

Next, we will calculate the double integral of the function $z = f(x,y) = x+y$, in the interval $0<x<2$, $0<y<2$. Before doing that, however, we produce a plot of the function:

```
-->deff('[z]=f(x,y)','z=x+y')                              //define z = f(x,y)
-->xx = [0:0.2:2]; yy = [0:0.2:2]; zz = feval(xx,yy,f);  //calculate z = f(x,y)
-->plot3d(xx,yy,zz)                                        //produce plot
```




To calculate the double integral using 10 sub-intervals in both x and y use:

```
-->DoubleIntegral(0,2,10,0,2,10,f)
 ans  =

    8.
```

Note: Notice that the plot shows a box extending between 0<x<2, 0<y<2, 0<z<4, whose volume is 2×2×4 = 16. Since the plane representing the function z = x+y divides the box into two halves, the volume representing the double integral will be half of the volume of the box, namely, 8.

The following calls to function *DoubleIntegral* for $0 < x < 1$ and $0 < y < 1$, use 5, 10, and 20 sub-intervals with $z = f(x,y) = x+y$:

```
-->DoubleIntegral(0,1,5,0,1,5,f)
 ans  = 1.

-->DoubleIntegral(0,1,10,0,1,10,f)
 ans  = 1.

-->DoubleIntegral(0,1,20,0,1,20,f)
 ans  = 1.
```

In the next exercise we use the function $z=f(x,y)=sin(x+y)$. First, we define the function and produce a plot for $0<x<2\pi$ and $0<y<2\pi$:

```
-->deff('[z]=f(x,y)','z=sin(x+y)')

-->xx = [0:%pi/10:2*%pi]; yy = xx; zz = feval(xx,yy,f);

-->plot3d(xx,yy,zz)
```




Due to the symmetry of the function above and below the xy plane, we suspect that the double integral will add to zero. A call to function *DoubleIntegral* in the interval under consideration confirms our suspicion:

```
-->DoubleIntegral(0,2*%pi,20,0,2*%pi,20,f)
 ans  = - 2.049E-15
```

Similar symmetry occurs for the function in the interval $0<x<2\pi$ and $0<y<2\pi$, the double integral on those intervals producing again a zero result:

```
-->DoubleIntegral(0,%pi,20,0,%pi,20,f)
 ans  = 3.137E-16
```

The following plot shows the function $z = sin(x+y)$ for the interval $0<x<1$ and $0<y<2$.

```
-->xx = [0:0.1:1]; yy = [0:0.1:2]; zz = feval(xx,yy,f);plot3d(xx,yy,zz)
```



The figure does not show the same symmetry of the surface above and below the x-y plane as in the two previous intervals. The integral of the function on this region produces indeed a non-zero value:

```
-->DoubleIntegral(0,1,20,0,2,20,f)
 ans  =  1.6079717
```

Double integrals transforming integration region into a rectangle

Suppose we are to calculate the double integral

$$I = \int_a^b \int_{g_1(x)}^{g_2(x)} f(x,y)dydx$$

in the region $R = \{g_1(x) \le y \le g_2(x),\ a \le x \le b\}$, illustrated in part [a] of the figure below.



[a] Original region R in the x-y plane [b] Transformed region in the x-z plane

We can use function *DoubleIntegral* to calculate the double integral indicated above if we transform the original region of integration *R* in the x-y plane into a rectangular region of integration through the coordinate transformation

$$y = [g_2(x)-g_1(x)]z + g_1(x).$$

The resulting rectangular region $R' = \{\ 0 \le z \le 1,\ a \le x \le b\ \}$ is illustrated in part [b] of the figure above. Function $f(x,y)$ gets transformed into function $\phi(x,z)$ through

$$\phi(x,y) = f(x,\ [g_2(x)-g_1(x)]z + g_1(x)) = \phi(x,z).$$

To put together the double integral in the *x-z* plane we need to replace the value *dy* with

$$dy = g_2(x)-g_1(x).$$

Thus, the double integral to be calculated is now

$$I = \int_a^b \int_{g_1(x)}^{g_2(x)} f(x,y)dydx = \int_a^b \int_0^1 \phi(x,z)\cdot[g_2(x)-g_1(x)]\cdot dzdx = \int_a^b \int_0^1 \psi(x,z)\cdot dzdx,$$

where

$$\psi(x,y) = \phi(x,z)\cdot[g_2(x)-g_1(x)] = f(x,\ [g_2(x)-g_1(x)]z + g_1(x))\cdot[g_2(x)-g_1(x)]$$

is the new integrand function.

As an example, to calculate the double integral

$$I = \int_0^3 \int_1^{\sqrt{x/3}} \exp(y^3) dy dx$$

we identify $g_1(x) = 1$, $g_2(x) = (x/3)^{1/2}$, and use the transformation $y = ((x/3)^{1/2}-1)z+1$, to obtain a new function $\phi(x,z) = \exp([((x/3)^{1/2}-1)z+1]^3)$. With $dy = ((x/3)^{1/2}-1)$, the double integral itself gets transformed to

$$I = \int_0^3 \int_0^1 \exp([(\sqrt{x/3}-1)z+1]^3) \cdot (\sqrt{x/3}-1) \cdot dz dx .$$

For the purpose of calculating the integral with function *DoubleIntegral* we need to define the function

$$\psi(x,z) = \exp([((x/3)^{1/2}-1)z+1]^3).((x/3)^{1/2}-1).$$

The SCILAB commands to calculate the double integral are

```
-->deff('[p]=psi(x,z)',...
-->'p=exp(((sqrt(x/3)-1)*z+1)^3)*(sqrt(x/3)-1)')

-->I=DoubleIntegral(0,3,100,0,1,100,psi)
 I  =  - 1.7189023
```

Simpson's 1/9 rule for double integrals

In this section we present an outline of the application of Simpson's rule to double integrals on a rectangular domain. The domain is divided as indicated in the figure below.



The function to be integrated is discretized to values $f_{ij} = f(x_i,y_j)$, $i=0,1,\ldots,n$, $j=0,1,\ldots,m$. The increments in x and y are Δx and Δy, respectively. Also, as with the single-variable Simpson's 1/3 rule, the values n and m must be even for the Simpson's 1/9 rule for double integrals.

Applying the Simpson's 1/3 rule to the nine points in the sub-region

$$R_{1,1} = \{x_0 \le x \le x_2,\ y_0 \le y \le y_2\}$$

results in the expression

$$\int_{x_0}^{x_2}\int_{y_0}^{y_2} f(x,y)dydx \approx \frac{1}{9}\Delta x \cdot \Delta y \cdot [\, f_{0,0} + f_{0,2} + f_{2,0} + f_{2,2} + 4\{ f_{0,1} + f_{1,0} + f_{1,2} + f_{2,1}\} + 16 f_{1,1}\,].$$

In general, the integral on the nine points contained in each of the 9-point sub-regions

$$R_{ij} = \{x_{i-1} \le x \le x_{i+1},\ y_{i-1} \le y \le y_{i+2}\},$$

for $i = 1,2,\ldots,n-1$, and $j = 1,2,\ldots,m-1$, is calculated as

$$S_{ij} = \frac{1}{9}\Delta x \cdot \Delta y \cdot [\, f_{i-1,0=i-1} + f_{i-1,i+1} + f_{i+1,i-1} + f_{i+1,i+1} + 4\{ f_{i-1,i} + f_{i,i-1} + f_{i,i+1} + f_{i+1,i}\} + 16 f_{i,i}\,].$$

To calculate the integral through the full region, i.e., in

$$R = \{x_0 \le x \le x_n,\ y_0 \le y \le y_m\},$$

one needs to calculate the $(n-1)(m-1)$ summations S_{ij} for $i = 1,2,\ldots,n-1$, and $j = 1,2,\ldots,m-1$, and add them all together. Thus, we can write

$$I = \int_{x_0}^{x_n}\int_{y_0}^{y_m} f(x,y)dydx \approx \sum_{i=1}^{n-1}\sum_{j=1}^{m-1} S_{ij},$$

where the sums S_{ij} were defined above.

The implementation of Simpson's 1/9 rule for double integrals is left as an exercise for the reader.

SCILAB function for calculating double integrals

SCILAB provides function *int2d* to calculate the integral of a function $z = f(x,y)$ over a region R defined by a number of triangles. The simplest call to the function is

[Int,er] = int2d(X,Y,f),

where *Int* is the value of the integral, *er* is the error involved in the calculation, *X* and *Y* are matrices of 3 rows and *n* columns representing the abscissas and ordinates, respectively, of the *n* triangles into which region *R* is divided.

As an example consider the integral of the function $f(x,y) = cos(x+y)$ in the region $R = \{0 < x < 5,\ 0 < y < 5\}$. The figure below shows two possible ways of triangulating the region.



In the first case we will define the matrices X and Y as

$$X = \begin{bmatrix} 0 & 0 \\ 5 & 5 \\ 5 & 0 \end{bmatrix}, \quad Y = \begin{bmatrix} 0 & 0 \\ 0 & 5 \\ 5 & 5 \end{bmatrix},$$

while in the second case they will be defined as

$$X = \begin{bmatrix} 0 & 5 \\ 5 & 5 \\ 0 & 0 \end{bmatrix}, \quad Y = \begin{bmatrix} 0 & 0 \\ 0 & 5 \\ 5 & 5 \end{bmatrix}.$$

The following SCILAB commands show the calculation of the double integral for the two triangulations shown above:

```
-->deff('z=f(x,y)','z=cos(x+y)')
```

Triangles ABC-ACD:

```
-->X = [0,0;5,5;5,0]; Y = [0,0;0,5;5,5];

-->[Int,er] =int2d(X,Y,f)
  er  = 1.839E-09
 Int  = .4063959
```

Triangles ABD-BCD:

```
-->X = [0,5;5,5;0,0]; Y = [0,0;0,5;5,5];
```



```
-->[Int,er] =int2d(X,Y,f)
 er  =     1.839E-09
 Int  =    .4063959
```

If the region of integration is a single triangle, you need to provide only the coordinates of that triangle to obtain and integral with function *int2d*. Consider the case of the double integral of the function $z = f(x,y) = x^2+y^2$ in the triangular region shown below.



The SCILAB commands to calculate this integral are as follows:

```
-->deff('[z]=f(x,y)','z=x^2+y^2')

-->X = [0;2;1]; Y =[0;0;1];

-->[Int,er] =int2d(X,Y,f)
 er  =  2.961E-16
 Int  = 1.3333333
```

Consider now the double integral of the function $z=f(x,y) = x + y$, in the region defined by $R = \{ 0<y<1-x^2,\ 0<x<1\}$. The figure below shows the region R with a triangulation based on a partition of the range 0<x<1 into 10 sub-intervals, x = 0, 0.1, 0.2, …, 1.0. The following SCILAB commands are used to generate the graph:

```
--> x = [0:0.1:1]; yT = 1-x.^2; yB = zeros(x);

--> xset('mark',-9,1)

--> plot2d([x',x',x'],[yT',yT',yB'],[1,-9,-9])

--> xtitle('triangulation of a region','x','y')

--> n = length(x)
 n  =   11.

--> for j = 1:n
-->     xpoly([x(j),x(j)],[yB(j),yT(j)],"lines")
--> end;


--> for j = 1:n-2
-->    xpoly([x(j),x(j+1)],[yB(j),yT(j+1)],"lines")
--> end;
```




The following *for…end* loops generate the matrices X and Y required for the application of function *int2d*:

```
--> X=[];Y=[];
--> for j = 1:n-2
-->     X = [X [x(j);x(j);x(j+1)]]; X = [X [x(j);x(j+1);x(j+1)]];
-->     Y = [Y [yB(j);yT(j);yT(j+1)]]; Y = [Y [yB(j);yT(j+1);yB(j+1)]];
-->end;

-->X = [X [x(n-1);x(n-1);x(n)]]; Y = [Y [yB(n-1);yT(n-1);yB(n)]];
```

The following statement defines the function to integrate:

```
-->deff('[z]=f(x,y)','z=x+y')
```

The integral is calculated as follows:

```
-->[Int,er] = int2d(X,Y,f)
 er  =

    1.283E-16
 Int  =

     .5147233
```

Application of double integrals

As their one-dimensional counterparts, double integrals can be used to calculate properties of areas and volumes. Some applications of double integrals are presented in this section.

Area properties with double integrals

For the calculation of area properties through the use of double integrals, refer to the following figure:



Area: since $dxdy$ represent an element of area in the plane, $dA = dxdy$, then,

$$A = \iint_R dA = \iint_R dydx .$$

Centroid: Let (x_c, y_c) represents the centroid of the region R, then

$$x_c = \frac{1}{A}\iint_R xdA \text{ , } y_c = \frac{1}{A}\iint_R ydA .$$

Areal moments of inertia: Let I_x, I_y , and I_0 , represent areal moments of inertia with respect to the x-,y- and z-axes, respectively, then

$$I_x = \iint_R y^2 dA \text{ , and } I_y = \iint_R x^2 dA \text{ , and } I_0 = I_x + I_y = \iint_R (x^2 + y^2) dA .$$

Radius of gyration: The radii of gyration of an area with respect to the x- ,y-, and z-axes are given by

$$k_x = \sqrt{\frac{I_y}{A}} \text{ , } k_y = \sqrt{\frac{I_x}{A}} \text{ , and } k_0 = \sqrt{\frac{I_0}{A}} ,$$

respectively.

Mass: if $f(x,y)$ represents the ***density (mass/unit area)*** of a body in the shape of the region of integration, then a differential of mass associated with the differential of area *dA = dy dx*, is *dM= f(x,y) dA = dM = f(x,y) dy dx*, and the mass of the region is

$$M = \iint_R dM = \iint_R f(x,y)dA = \iint_R f(x,y)dydx .$$

Center of mass: The center of mass of the region R has coordinates, x_c, y_c , given by:

$$x_c = \frac{1}{M}\iint_R xf(x,y)dA \quad , \quad y_c = \frac{1}{M}\iint_R yf(x,y)dA .$$

Moments of inertia of the mass: The moments of inertia I_x and I_y of the mass represented by region R about the x and y axes, respectively, are

$$I_x = \iint_R y^2 f(x,y)dA \text{ , and } I_y = \iint_R x^2 f(x,y)dA ,$$

and the polar moment of inertia (i.e., moment of inertia with respect to the origin, or, more correctly, with respect to the z-axis) is

$$I_0 = I_x + I_y = \iint_R (x^2 + y^2) f(x,y)dA .$$

Radius of gyration of the mass: The radii of gyration of the mass represented by region R with respect to the x- ,y-, and z-axes are given by

$$k_x = \sqrt{\frac{I_y}{M}} \quad , \quad k_y = \sqrt{\frac{I_x}{M}} \text{ , and } k_0 = \sqrt{\frac{I_0}{M}} ,$$

respectively.

Volume: the volume between the x-y plane and the surface *z = f(x,y)* is provided by the double integral,

$$V = \iint_R f(x,y)dA = \iint_R f(x,y)dydx .$$

Calculation of areal properties in Cartesian coordinates

For the quarter of circle defined by the region R ={ $x^2+y^2 = a^2$, $0<x<a$}, for *a = 2,* find the area, the centroid, and the areal moments of inertia with respect to the *x-, y-,* and *Z*-axes in a Cartesian coordinate system (*x,y,Z*). [**Note**: *Z* is used here instead of *z* to avoid confusing the *Z* coordinate of the Cartesian system with the *z* variable used in a transformation that converts the original region into a rectangular region as shown below].

The region of integration can be described as $R = \{ 0 < y < (a^2-x^2)^{1/2}, 0 < x < a\}$. To calculate the double integrals using the user-defined function *DoubleIntegral* we need to transform the

region *R* in the x-y plane into a rectangular region *R' = {0<z<1, 0<x<a}* in the *z-x* plane, by using the transformation

$$y = z\cdot(a^2-x^2)^{1/2},$$

$$dy = (a^2-x^2)^{1/2}\,dz.$$

The function to be integrated *f(x,y)* in the *x-y* plane gets transformed into

$$\phi(x,z) = f(x,z\cdot(a^2-x^2)^{1/2}),$$

thus, we will calculate

$$\int_0^a \int_0^{\sqrt{a^2-x^2}} f(x,y)dydx = \int_0^a \int_0^1 \phi(x, z\sqrt{a^2-x^2})\cdot\sqrt{a^2-x^2}\cdot dzdx.$$

The area is calculated by taking *f(x,y) = 1.0* as follows:

$$A = \int_0^a \int_0^{\sqrt{a^2-x^2}} dydx = \int_0^a \int_0^1 \sqrt{a^2-x^2}\cdot dzdx.$$

Using SCILAB and the user-defined function *DoubleIntegral* we find the following approximation to the area:

```
-->deff('[w]=g(x,y)','w=sqrt(4-x^2)')

-->A = DoubleIntegral(0,2,1000,0,1,50,g)
 A  =  3.1415555
```

The coordinates of the centroid are calculated using $x_c = M_y/A$, and $y_c = M_x/A$, with

$$M_y = \iint_R x dA = \int_0^a \int_0^{\sqrt{a^2-x^2}} xdydx = \int_0^a \int_0^1 x\sqrt{a^2-x^2}\,dzdx,$$

and

$$M_x = \iint_R y dA = \int_0^a \int_0^{\sqrt{a^2-x^2}} ydydx = \int_0^a \int_0^1 (z\sqrt{a^2-x^2})(\sqrt{a^2-x^2})dzdx =$$

$$\int_0^a \int_0^1 z(a^2-x^2)dzdx.$$

Using SCILAB we calculate the centroidal coordinates as

```
-->deff('[w]=my(x,z)','w=x*sqrt(4-x^2)')

-->deff('[w]=mx(x,z)','w=z*(4-x^2)')

-->My = DoubleIntegral(0,2,1000,0,1,50,my)
 My  = 2.6665916

-->Mx = DoubleIntegral(0,2,1000,0,1,50,mx)
Mx  = 2.666666
```



```
-->xc=My/A, yc=Mx/A
 xc  = .8488125
 yc  = .8488362
```

The moments of inertia of the area are

$$I_y = \iint_R x^2 dA = \int_0^a \int_0^{\sqrt{a^2-x^2}} x^2 dydx = \int_0^a \int_0^1 x^2 \sqrt{a^2 - x^2}\, dzdx,$$

and

$$I_x = \iint_R y^2 dA = \int_0^a \int_0^{\sqrt{a^2-x^2}} y^2 dydx = \int_0^a \int_0^1 (z\sqrt{a^2 - x^2})^2 (\sqrt{a^2 - x^2}) dzdx =$$

$$\int_0^a \int_0^1 z^2 (a^2 - x^2)^{3/2} dzdx.$$

Using SCILAB:

```
-->deff('[w]=ix(x,z)','w=x^2*sqrt(4-x^2)')

-->deff('[w]=iy(x,z)','w=z^2*(4-x^2)^(3/2)')

-->Ix=DoubleIntegral(0,2,1000,0,1,50,ix)
 Ix  = 3.1414439

-->Iy=DoubleIntegral(0,2,1000,0,1,50,iy)
 Iy  = 3.142221
```

The radii of gyration are $k_x = (I_x/A)^{1/2}$ and $k_y = (I_y/A)^{1/2}$, i.e.,

```
-->kx=sqrt(Ix/A), ky=sqrt(Iy/A)
 kx  =  .9999823
 ky  = 1.0001059
```

Change of variables in double integrals

Let R is a region in the x-y plane. Suppose that we want to evaluate the double integral,

$$\iint_R f(x,y) dA_{x,y} = \iint_R f(x,y) dydx$$

,

by changing variables through the coordinate transformation $x = x(u,v)$, $y = y(u,v)$. The transformation of coordinates maps the region R in the x-y plane into a region R* in the u-v plane. The double integral in terms of the variables u and v incorporates a quantity known as the Jacobian of the transformation. We define the *Jacobian* of the coordinate transformation as the determinant

$$J = \frac{\partial(x,y)}{\partial(u,v)} = \begin{vmatrix} \frac{\partial x}{\partial u} & \frac{\partial x}{\partial v} \\ \frac{\partial y}{\partial u} & \frac{\partial y}{\partial v} \end{vmatrix}.$$

With this definition, then we can write

$$\iint_R f(x,y)\,dA_{x,y} = \iint_R f(x,y)\,dy\,dx = \iint_{R'} f[x(u,v),y(u,v)]\,dA_{u,v} = \iint_{R'} f[x(u,v),y(u,v)]\,|J|\,du\,dv$$

.

The latter equation suggests that the differential of area in the u-v plane is:

$$dA_{u,v} = |J|\,du\,dv$$

.

Jacobian functions can be obtained for any coordinate transformations in systems of three or more coordinates. Thus, a generalized definition of the Jacobian corresponding to the transformation $(x_1, x_2, \ldots, x_n) \rightarrow (\phi_1, \phi_2, \ldots, \phi_n)$, is

$$J = J\left(\frac{x_1, x_2, \ldots, x_n}{\phi_1, \phi_2, \ldots, \phi_n}\right) = \begin{vmatrix} \frac{\partial x_1}{\partial \phi_1} & \frac{\partial x_1}{\partial \phi_2} & \cdots & \frac{\partial x_1}{\partial \phi_n} \\ \frac{\partial x_2}{\partial \phi_1} & \frac{\partial x_2}{\partial \phi_2} & \cdots & \frac{\partial x_2}{\partial \phi_n} \\ \vdots & \vdots & \ddots & \vdots \\ \frac{\partial x_n}{\partial \phi_1} & \frac{\partial x_2}{\partial \phi_n} & \cdots & \frac{\partial x_n}{\partial \phi_n} \end{vmatrix}.$$

Area calculation using double integrals with polar coordinates

Suppose we want to find the area of a circle of radius R using polar coordinates. The transformations are:

$$x := (r, \theta) \rightarrow r\cos(\theta)$$

$$y := (r, \theta) \rightarrow r\sin(\theta)$$

The Jacobian of the transformation (x,y) to (r, θ) is:

$$\begin{bmatrix} \cos(\theta) & -r\sin(\theta) \\ \sin(\theta) & r\cos(\theta) \end{bmatrix}$$

$$J := r$$

Therefore, the differential of area $dA_{r,\theta}$ in the r-θ plane is $dA_{r,\theta} = |J|dr\,d\theta = r\,dr\,d\theta$. The shaded element in the figure below is a typical area differential in polar coordinates. Although it is not shaped as a rectangle, because we are dealing with small quantities, we can approximate it to a rectangle of dimensions $r\,d\theta$ and dr, from which the value $dA_{r,\theta} = r\,dr\,d\theta$, follows.



Since we are trying to obtain the area of a region in x-y (a circle of radius R), the function to integrate is $f(x, y) = 1.0$, which transforms to f[$x(r, \theta), y(r, \theta)$] = 1.0. The region in the x-y plane is the circle itself, R = { $x^2 + y^2 \le R^2$ }, which in polar coordinates is mapped as R* = { $0 \le r, r \le R$, and $0 \le \theta, \theta \le 2\pi$ }. Therefore, the integral $\int\int 1.0\, dy\, dx$ over R, becomes:

$$\int_0^{2\pi} \int_0^R J\, dr\, d\theta = R^2 \pi$$

Double integral applications in polar coordinates

Some double integral applications using polar coordinates are shown below:

Area: since *dxdy* represent an element of area in the plane, $dA = dxdy$, then,

$$A = \iint_{R^*} dA_{r,\theta} = \iint_{R^*} r dr d\theta .$$

Centroid: Let (r_c, θ_c) represents the centroid of the region R*, then

$$r_c = \frac{1}{A} \iint_R r dA_{r,\theta} \quad , \quad \theta_c = \frac{1}{A} \iint_{R^*} \theta dA .$$

Polar areal moment of inertia: The moment of inertia I_0 , represents the so-called polar of inertia, i.e., with respect to the z-axis, is

$$I_0 = \iint_{R^*} r^2 dA_{r,\theta} .$$

Radius of gyration: The radii of gyration of an area with respect to the z-axis is given by

$$k_0 = \sqrt{\frac{I_0}{A}} .$$

Mass: if $f(r, \theta)$ represents the *density (mass/unit area)* of a body in the shape of the region of integration, then a differential of mass associated with the differential of area $dA_{r,\theta} = rdr\, d\theta$, is dM= $f(r, \theta)$ $dA_{r,\theta}$ = f(r, θ) $rdr\, d\theta$, , and the mass of the region is

$$M = \iint_{R^*} dM = \iint_{R^*} f(r,\theta)\,dA_{r,\theta} = \iint_{R^*} f(r,\theta)\,r\,dr\,d\theta .$$

Center of mass: The center of mass of the region R has coordinates, r_c, θ_c , given by:

$$r_c = \frac{1}{M}\iint_{R^*} r f(r,\theta)\,dA_{r,\theta} \quad , \quad \theta_c = \frac{1}{M}\iint_{R^*} \theta f(r,\theta)\,dA_{r,\theta} .$$

Moments of inertia of the mass: The polar moment of inertia (i.e., moment of inertia with respect to the origin) is

$$I_0 = \iint_{R^*} r^2 f(r,\theta)\,dA_{r,\theta} .$$

Radius of gyration of the mass: The radii of gyration of the mass represented by region R with respect to the z-axes is given by

$$k_0 = \sqrt{\frac{I_0}{M}} .$$

Volume: the volume between the x-y plane and the surface $z = f(r,\theta)$ is provided by the double integral,

$$V = \iint_{R^*} f(r,\theta)\,dA = \iint_{R^*} f(r,\theta)\,r\,dr\,d\theta .$$

Calculation of areal properties in polar coordinates

Calculate the area, centroid, polar moment of inertia, and radius of gyration of the region defined by $R^* = \{0<r<\sin(2\theta),\ 0<\theta<\frac{\pi}{2}\}$.

A plot of the region follows:



The area is calculated as

$$A = \int_0^{\pi/2}\int_0^{\sin(2\theta)} r\,dr\,d\theta .$$

The region of integration in the *r-θ* plane (not the x-y plane using polar coordinates, but a plane where the abscissas are represented by *r* and the ordinates by *θ*) will be an irregularly shaped region *R* which can be transformed into a rectangular region R' in a *r-z* plane. The transformation to use is *r = z sin(2θ), dr = sin(2θ)dz*. The area is calculated now as

$$A = \int_0^{\pi/2}\int_0^1 z\sin(2\theta)\sin(2\theta)dzd\theta = \int_0^{\pi/2}\int_0^1 z\sin^2(2\theta)dzd\theta\ .$$

Using SCILAB:

```
-->deff('[w]=g(z,th)','w=z*(sin(2*th))^2')

-->A = DoubleIntegral(0,1,100,0,%pi/2,100,g)
 A  =  .3926991
```

The coordinates of the centroid are $r_c = (1/A)\iint_A r^2 dr d\theta$, $\theta_c = (1/A)\iint_A \theta\, r dr d\theta$. Using the transformation *r = z sin(2θ), dr = sin(2θ)dz*, the integrals are calculated as

$$r_c = \frac{1}{A}\int_0^{\pi/2}\int_0^1 z^2\sin^3(2\theta)dzd\theta, \quad \theta_c = \frac{1}{A}\int_0^{\pi/2}\int_0^1 z\theta\sin^2(2\theta)dzd\theta.$$

Using SCILAB:

```
-->deff('[w]=mr(z,th)','w=z^2*(sin(2*th))^3')

-->deff('[w]=mt(z,th)','w=z*th*(sin(2*th))^2')

-->rc = DoubleIntegral(0,1,100,0,%pi/2,100,mr)/A
 rc  = .5659125

-->thc = DoubleIntegral(0,1,100,0,%pi/2,100,mt)/A
 thc  = .7853982
```

The polar moment of inertia is calculated with $I_c = \iint_A r^3 dr d\theta$. With the coordinate transformation *r = z sin(2θ), dr = sin(2θ)dz*, the integral to calculate is

$$I_0 = \frac{1}{A}\int_0^{\pi/2}\int_0^1 z^3\sin^4(2\theta)dzd\theta.$$

Using SCILAB

```
-->deff('[w]=i0(z,th)','w=z^3*(sin(2*th))^4')

-->I0 = DoubleIntegral(0,1,100,0,%pi/2,100,i0)
 I0  = .1472769
```

The radius of gyration about the origin is calculated as $k_0 = (I_o/A)^{1/2}$, i.e.,

```
-->k0=sqrt(I0/A)
 k0  = .6124031
```


A final note on the numerical calculation of double integrals

All the examples of applications of numerical calculation of double integrals presented here have been calculated using the user-defined function *DoubleIntegral* with coordinate transformations that convert the original integration region into a rectangular region. It should be indicated that the approach followed in function *DoubleIntegral* is based on approximating the volume represented by the integral $\iint_A f(x,y)\,dxdy$ by the sum of parallelepiped elements whose base has an area $\Delta x \cdot \Delta y$, and whose height is $f(x_i,y_j)$. Function *DoubleIntegral* uses a fixed and uniform grid to approximate the integral. The error incurred in the approximation is of first order, i.e., $O(\Delta x \cdot \Delta y)$. Thus, you may have to use a very fine grid to obtain reasonable results with this function.

Improved results for double integrals can be obtained by coding Simpson's 1/9 rule as indicated in a previous section, or by using SCILAB function *int2d*. Function *int2d* uses an adaptive algorithm by which regions in which the function varies rapidly are divided into finer grids and the integral recalculated until a convergence is achieved. Adaptive methods can also be used to calculate integrals of one variable. Such algorithms are incorporated in many of SCILAB's predefined integral functions such as *intg* or *integrate*.

Exercises

For problems [1] through [10], use the user-defined function *Sumint* to produce the approximation to the integral of *f(x)* between the limits $a<x<b$ using *n* subintervals. Calculate the integrals using (a) the left sum, (b) the middle sum, and (c) the right sum.

[1] $f(x) = 1/(1+ \ln(x))$, $a = 1$, $b = 5$, $n = 30$

[2] $f(x) = 1 + x^2 + x^3 + 1/x$, $a = 1$, $b= 3.5$, $n = 90$

[3] $f(x) = \exp(-x/5)/(1+x^2)$, $a = -2$, $b = 2$, $n = 50$

[4] $f(x) = (x^2+1)^{1/2}$, $a = -1$, $b = 1$, $n = 20$

[5] $f(x) = (x^2+1)^{1/3}+2.5(x^2+1)+1.5$, $a = 0$, $b = 5$, $n = 60$

[6] $f(x) = \sinh(x)/(1+x^3)$, $a = 0$, $b = 3$, $n = 40$

[7] $f(x) = \exp(-x^2/2)/(2\pi)^{1/2}$, $a = -4$, $b = 4$, $n = 100$

[8] $f(x) = 2.5(1+x^2+x^3/3)\cos(3x^2+2)$, $a = 0$, $b = 2\pi$, $n = 25$

[9] $f(x) = 4.2x^{1/3} + 3.245x^{1/2} + 1.4142$, $a = 0.25$, $b = 1.25$, $n = 20$

[10] $f(x) = 0.33 \ln((x+1)/(x^2+5x+2))$, $a = 0$, $b = 20$, $n = 40$

[11] - [20] Repeat problems [1] through [10] using (a) the trapezoidal rule through SCILAB function *inttrap*; (b) Simpson's 1/3 rule through function *simpson13* if the number of

subintervals is even, and/or Simpson's 3/8 rule through function *simpson38* if the number of subintervals is a multiple of 3.

[21] - [30] Repeat problems [1] through [10] using the Newton-Cotes formulas with (a) $n=3$, (b) $n=4$, (c) $n=5$, (d) $n=6$, (e) $n=7$.

[31] - [40] Repeat problems [1] through [10] using (a) SCILAB function *integrate*; (b) integration by splines using SCILAB function *intslpin*; (c) *SCILAB* function *intg*.

In problems [41] through [50], given a function of a complex variable $w = F(z)$, with $z = x + iy$, identify the real and imaginary components $\Phi(x,y)$ and $\Psi(x,y)$ such that $F(z) = \Phi(x,y) + i\,\Psi(x,y)$. Also, produce contour plots of the functions $\Phi(x,y)$ and $\Psi(x,y)$.

[41] $F(z) = 1/z$

[42] $F(z) = z + 1/z$

[43] $F(z) = exp(z)$

[44] $F(z) = z^3$

[45] $F(z) = z^{1/2}$

[46]-[50]. For problems [46] through [50], integrate the function of a complex variable from problems [41] through [45] using SCILAB function *intl* along the straight line joining the points z_1 and z_2, where (a) $z_1 = 3,\ z_2 = 2-i$; (b) $z_1 = -5+2i,\ z_2 = 2-2i$; (c) $z_1 = 3i,\ z_2 = 5+2i$; (d) $z_1 = (1+i)/2,\ z_2 = 3+\sqrt{2}i$; (e) $z_1 = i/2$, $z_2 = -i/2$.

[51]-[55]. For problems [51] through [55] use function *intl* to calculate the integral of the function of a complex variable from problems [41] through [45] along the curve defined by $z = z_0 + r\cdot e^{i\theta}$, within the limits θ_1 and θ_2, using the following values: (a) $z_0 = 2+3i,\ r = 1,\ \theta_1 = 0,\ \theta_2 = \pi/4$; (b) $z_0 = 1+i,\ r = 0.2,\ \theta_1 = -\pi/2,\ \theta_2 = \pi/2$; (c) $z_0 = 0,\ r = 1,\ \theta_1 = \pi,\ \theta_2 = 2\pi$; (d) $z_0 = -1-2i,\ r = 0.75,\ \theta_1 = 0,\ \theta_2 = 3\pi/4$; (e) $z_0 = 3,\ r = 1,\ \theta_1 = -\pi/2,\ \theta_2 = \pi$.

For problems [56]-[60] use the user-defined function *DoubleIntegral* to calculate the double integral of function $f(x,y)$ in the rectangular domain $a < x < b$, $c < y < d$.

[56]. $f(x,y) = x^2+y^2,\ a = -2,\ b = 2,\ c = -1,\ d = 1.$

[57]. $f(x,y) = x \sin y + y \sin x,\ a = -\pi/2,\ b = \pi/2,\ c = 0,\ d = \pi.$

[58]. $f(x,y) = 2xy(x^2+y^2+1)^{1/2},\ a = 0,\ b = 2,\ c = 0,\ d = 2.$

[59]. $f(x,y) = x\ ln(y/x) + y\ ln(x/y),\ a = 1,\ b = 2,\ c = 0.5,\ d = 0.75.$

[60]. $f(x,y) = 1-exp(-(x^2+y^2)/25),\ a = -2,\ b = 2,\ c = -2,\ d = 2.$

[61]-[65]. Repeat problems [56] through [60] using SCILAB function *int2d* by dividing the rectangular region of integration into two triangles.

[66]-[70]. For problems [66] through [70] use function *int2d* to calculate the double integral of the functions in problems [56] through [60] on the integration region defined by: (a) $R = \{-1 < x < 1,\ 0 < y < (1-x^2)\}$; (b) $R = \{ 0<x< 1,\ x^2 < y < x\}$; (c) $R = \{-1 < x < 1,\ 0 < y < 1 - x^2\}$; (d) $R = \{-3 < x < 3,\ |x| < y < 3\}$; (e) $R = \{-4 < x < 4,\ (25-x^2)^{1/2} < y < (16-x^2)^{1/2}\}$.

[71].Determine the area under the curve $y = f(x)$ limited by the values $x = a$ and $x = b$:
(a) $f(x) = x^2$, $a=2$, $b=5$ (b) $f(x) = ln(x)$, $a = 2$, $b = 3$
(c) $f(x) = 1/(1+x^2)$, $a = 0$, $b =1$ (d) $f(x) = (1-x^2)^{1/2}$, $a = -0.5$, $b = 0.5$

[2]. Determine the area between the curves $y = f_1(x)$ and $y = f_2(x)$ and the values $x = a$ and $y = b$. Plot the region to be integrated:
(a) $f_1(x) = x^{1/2}$, $f_2(x) = x^2$, $a = 0$, $b = 1$ (b) $f_1(x) = (1-x^2)^{1/2}$, $f_2(x) = -(1-x^2)^{1/2}$, $a = -1$, $b = 1$
(c) $f_1(x) = 10(1-x/5)$, $f_2(x) = 2$, $a =0$, $b =3$ (d) $f_1(x) = (1-x^2)^{1/2}$, $f_2(x) = x$, $a = 0$, $b =$ *point of intersection*

[73]. Determine the center of mass (centroid) of the figure limited by the x-axis and the curve $y = f(x)$ between the values $x = a$ and $x = b$ for the functions of problem [71].

[74]. Determine the volume of the solid of revolution generated by the rotation of the curve $y = f(x)$ between the values $x = a$ and $x = b$ about the x axis for the functions of problem [71].

[75]. Determine the moments of inertia and radii of gyration ($k_x = (I_x/A)^{1/2}$, $k_y = (I_y/A)^{1/2}$) with respect to the x-axis, y-axis, and the origin of the area limited by the x-axis and the curve $y = f(x)$ between the values $x = a$ and $x = b$, for the functions of problem [71].

[76]. For the figures in problem [75] determine the moments of inertia and radii of gyration with respect the centroidal axes and the centroid. Use the parallel axes moment.

[77]. Determine the area defined by $r = f(\theta)$ and the radii $r = \theta_1$ and $r = \theta_2$. Plot the region using polar coordinates:
(a) $f(\theta) = 3 \cos \theta$, $\theta_1 = 0$, $\theta_2 = \pi/2$ (b) $f(\theta) = 1-\sin \theta$, $\theta_1 = -\pi/2$, $\theta_2 = \pi/2$
(c) $f(\theta) = 1/(1+\cos \theta)$, $\theta_1 = 0$, $\theta_2 = \pi/2$ (d) $f(\theta) = \theta$, $\theta_1 = 0$, $\theta_2 = \pi$

[78]. Calculation of discharge in a pipe. Calculate the discharge in a pipe of radius R = 2.5 ft, if the velocity distribution in the pipe cross-section is given by:
(a) $v(r) = v_0(1-r/R)$ (b) $v(r) = v_0[1-(r/R)^3]$ (c) $v(r) = v_0[1-(r/R)^{1/2}]$
The value of $v_o = 3.5$ *fps*.

[79]. Integrate the function f(x,y,z) along curve *C* using the suggested variable substitutions:
(a) $f(x,y,z) = 1-x^2-y^2$, $x(s) = sin(s)$, $y(s) = cos(s)$, $z(s) = 0$, $s = 0..\pi$
(b) $f(x,y,z) = xyz$, $x(s) = 2 sin(s)$, $y(s) = 3 cos(s/2)$, $z(s) = s$, $s = 0..\pi$
(c) $f(x,y,z) = sin\ x\ cos\ y$, $x(s) = s$, $y(s) = s^2/2$, $z(s) = 0$, $s = 0..2$
(d) $f(x,y,z) = 1/(x^2+y^2+1)$, $x(s) = s^2$, $y(s) = ln\ s$, $z(s) = 0$, $s = 1..10$

[80]. Calculate the line integral for the differential $f(x,y,z)dx + g(x,y,z)dy + h(x,y,z)dz$ along the curve C specified:
(a) $f(x,y,z)=x+y$, $g(x,y,z) =xy$, $h(x,y,z) = z^2$, $C:\{ y = x+2, z = x-1, 0<x<5 \}$
(b) $f(x,y,z)=x+yz$, $g(x,y,z) =y+xz$, $h(x,y,z) =z+xy$, $C:\{x^2+y^2=1, 0<x<1, y>0, z = 2\}$
(c) $f(x,y,z)=sin\ x$, $g(x,y,z) =sin\ y$, $h(x,y,z) =sin\ z$, $C:\{z = x+1, y=2, 0<x<2\}$
(d) $f(x,y,z)=x\ exp(y)$, $g(x,y,z) = exp(z)$, $h(x,y,z) =y^2$, $C: \{x = 2, y = z^2+1, 0< z<1\}$

[81]. Check if the differential $f(x,y,z)dx + g(x,y,z)dy + h(x,y,z)dz$ is an exact differential, and calculate the line integral between the selected points A and B:
(a) $f(x,y,z)=2x$, $g(x,y,z) =2y$, $h(x,y,z) = 2z$, $A: (0,0,0)$, $B(1,-1,2)$
(b) $f(x,y,z)=z\ cos(x)exp(y)$, $g(x,y,z) = z^2\ sin(x)\ exp(y)$, $h(x,y,z) = 2\ z\ sin(x)\ exp(y)$, $A:(1,1,1)$, $B:(2,2,2)$

(c) $f(x,y,z)= g(x,y,z) = h(x,y,z) =2(x+y+z)$, $A:(2,-2,1)$, $B:(5,5,-2)$
(d) $f(x,y,z)=y\cos(xy)$, $g(x,y,z) = x\cos(xy)$, $h(x,y,z) =0$, $A:(-1,1,0)$, $B:(2,2,2)$

[82]. Calculate the following double integrals using function *DoubleIntegral*:

$$(a)\int_0^2\int_{-x}^{x}(x+xy)dydx \qquad (b)\int_{-2}^{2}\int_{-y}^{0}(x+xy)dxdy \qquad (c)\int_0^2\int_{-x}^{x}\sqrt{x}dxdy \qquad (d)\int_0^2\int_{-2}^{\xi^2-1}\xi\eta d\eta d\xi$$

[83]. Using double integrals, calculate the area, coordinates of the centroid, areal moments of inertia with respect to the x- and y-axes and with respect to the origin of the area described by the region R. Plot the region of interest:
(a) $R: \{0<y<x^2+1,\ 0<x<1\}$ (b) $R: \{-1 < y < (1-x^2),\ 0 < x < 1\}$
(c) $R: \{0<y<1-x,\ 0 < x < 1\}$ (d) $R: \{0 < x < (1-y^2)^{1/2},\ -1 < y < 1\}$

[84]. Calculate the mass, coordinates of the center of mass, mass moments of inertia with respect to the x- and y-axes and with respect to the origin of the area described for the regions R from problem [83] if the corresponding areal densities are given by f(x,y):
(a) $f(x,y) = x$ (b) $f(x,y) = (x+1)^{1/2}$ (c) $f(x,y) = \sin(x)$ (d) $f(x,y) = 1/(x^2+y^2)$

[85]. Determine the volume of the three-dimensional region described by the surface $z = f(x,y)$ and limited by the region R:
(a) $f(x,y) = x^2+y$, $R:\{x = -2..2,\ y = -2..x^{1/2}\}$
(b) $f(x,y) = \sin(x)\cos(y)$, $R:\{x = -2\pi..2\pi,\ y = -2\pi..\sin(x)\}$
(c) $f(x,y) = \exp(-0.05*x)*\cos(y+2)$, $R:\{x = 0..5,\ y = 0..25\}$
(d) $f(x,y) = |\ln(x^2+y^2+10)|$, $R:\{x = -2..2,\ y = -2..2\}$

[86]. Calculate the area, coordinates of the centroid, areal moments of inertia with respect to the x- and y-axes and with respect to the origin of the area described by the region R (polar coordinates):
(a) $R:\{0 < r < 3\theta,\ 0 < \theta < \pi\}$ (b) $R:\{0 < r < \sin\theta,\ 0 < \theta < \pi/2\}$
(c) $R:\{0 < r < (\cos\theta - 1),\ 0 < \theta < \pi\}$ (d) $R:\{0 < r < 1/(1+\cos\theta),\ -\pi/2 < \theta < \pi/2\}$

[87]. Calculate the mass, coordinates of the center of mass, mass moments of inertia with respect to the x- and y-axes and with respect to the origin of the area described by the regions R in problem [11], if the corresponding areal densities are $f(r,\theta)$:
(a) $f(r,\theta)=2.5$ (b) $f(r,\theta) = \sin(\theta)$ (c) $f(r,\theta) = 1-\sin(\theta)$ (d) $f(r,\theta) = |\theta|$

[88]. Determine the volume of the three-dimensional region described by the surface $z = f(r,\theta)$, as shown below, and limited by the region R (polar coordinates) from problem [86]:
(a) $f(r,\theta)=2.5\exp(-\theta)$ (b) $f(r,\theta) = \sin^2(\theta)$ (c) $f(r,\theta) = 1+\theta$ (d) $f(r,\theta) = (1+\sin(\theta))^{1/2}$

[89]. The outline for the calculation of double integrals using Simpson's 1/9 rule was presented earlier. Write a SCILAB function that takes as input the limits of a rectangular region in the x-y plane, $R = \{x_0 < x < x_n,\ y_0 < y < y_m\}$, the (even) values n and m, and the function $f(x,y)$, and returns the approximation to the double integral

$$I = \int_{x_0}^{x_n}\int_{y_0}^{y_m} f(x,y)dydx.$$

[90]. Use the function for calculating double integrals through Simpson's 1/9 rule developed in problem [89] to solve problem [82].

[91]. Use the function for calculating double integrals through Simpson's 1/9 rule developed in problem [89] to solve problem [83].

[92]. Use the function for calculating double integrals through Simpson's 1/9 rule developed in problem [89] to solve problem [84].

[93]. Use the function for calculating double integrals through Simpson's 1/9 rule developed in problem [89] to solve problem [85].

[94]. The table below shows the flow velocity, *v(fps)*, and suspended sediment concentration, *C(mg/l)*, as functions of the distance from the channel bed, *y(ft),* measured at a cross-section of a *800*-ft-wide (*b = 800 ft*), *7.8*-ft-deep (*h = 7.8 ft*) river cross-section that can be approximated by a rectangle.

y(ft)	*v(ft/s)*	*C(mg/l)*
0.7	*4.30*	*411*
0.9	*4.50*	*380*
1.2	*4.64*	*305*
1.4	*4.77*	*299*
1.7	*4.83*	*277*
2.2	*5.12*	*238*
2.7	*5.30*	*217*
2.9	*5.40*	*211*
3.2	*5.42*	*196*
3.4	*5.42*	*188*
3.7	*5.50*	*184*
4.2	*5.60*	*165*
4.8	*5.60*	*148*
5.8	*5.70*	*130*
6.8	*5.95*	*80*

The flow discharge, *Q(cfs),* is defined by the integral

$$Q = \int_A v(y)dA = \int_0^h v(y)\cdot b dy,$$

with the mean flow velocity calculated as $V = Q/A$, where $A = b \cdot h$ being the cross-sectional area.
The total flux of suspended sediment through the cross-section is defined as

$$Q_s = \int_A v(y)\cdot C(y)\cdot dA = \int_0^h v(y)\cdot C(y)\cdot b dy,$$

with the *flux-averaged concentration* given by $C_f = Q_s/Q$.

The terms used in the definitions above are illustrated in the figure below:



(a) Plot the velocity profile *v(y)-vs-y* and the concentration profile *C(y)-vs-y*.
(b) Determine the flow discharge, mean velocity, sediment flux, and flux averaged concentration for the data given using integration by splines.

REFERENCES (for all SCILAB documents at InfoClearinghouse.com)

Abramowitz, M. and I.A. Stegun (editors), 1965,"*Handbook of Mathematical Functions with Formulas, Graphs, and Mathematical Tables*," Dover Publications, Inc., New York.

Arora, J.S., 1985, "*Introduction to Optimum Design*," Class notes, The University of Iowa, Iowa City, Iowa.

Asian Institute of Technology, 1969, "*Hydraulic Laboratory Manual*," AIT - Bangkok, Thailand.

Berge, P., Y. Pomeau, and C. Vidal, 1984,"*Order within chaos - Towards a deterministic approach to turbulence*," John Wiley & Sons, New York.

Bras, R.L. and I. Rodriguez-Iturbe, 1985,"*Random Functions and Hydrology*," Addison-Wesley Publishing Company, Reading, Massachussetts.

Brogan, W.L., 1974,"*Modern Control Theory*," QPI series, Quantum Publisher Incorporated, New York.

Browne, M., 1999, "*Schaum's Outline of Theory and Problems of Physics for Engineering and Science*," Schaum's outlines, McGraw-Hill, New York.

Farlow, Stanley J., 1982, "*Partial Differential Equations for Scientists and Engineers*," Dover Publications Inc., New York.

Friedman, B., 1956 (reissued 1990), "*Principles and Techniques of Applied Mathematics*," Dover Publications Inc., New York.

Gomez, C. (editor), 1999, "*Engineering and Scientific Computing with Scilab*," Birkhäuser, Boston.

Gullberg, J., 1997, "*Mathematics - From the Birth of Numbers*," W. W. Norton & Company, New York.

Harman, T.L., J. Dabney, and N. Richert, 2000, "*Advanced Engineering Mathematics with MATLAB® - Second edition*," Brooks/Cole - Thompson Learning, Australia.

Harris, J.W., and H. Stocker, 1998, "*Handbook of Mathematics and Computational Science*," Springer, New York.

Hsu, H.P., 1984, "*Applied Fourier Analysis*," Harcourt Brace Jovanovich College Outline Series, Harcourt Brace Jovanovich, Publishers, San Diego.

Journel, A.G., 1989, "*Fundamentals of Geostatistics in Five Lessons*," Short Course Presented at the 28th International Geological Congress, Washington, D.C., American Geophysical Union, Washington, D.C.

Julien, P.Y., 1998,"*Erosion and Sedimentation*," Cambridge University Press, Cambridge CB2 2RU, U.K.

Keener, J.P., 1988, "*Principles of Applied Mathematics - Transformation and Approximation*," Addison-Wesley Publishing Company, Redwood City, California.

Kitanidis, P.K., 1997,"*Introduction to Geostatistics - Applications in Hydogeology*," Cambridge University Press, Cambridge CB2 2RU, U.K.

Koch, G.S., Jr., and R. F. Link, 1971, "*Statistical Analysis of Geological Data - Volumes I and II*," Dover Publications, Inc., New York.

Korn, G.A. and T.M. Korn, 1968, "*Mathematical Handbook for Scientists and Engineers*," Dover Publications, Inc., New York.

Kottegoda, N. T., and R. Rosso, 1997, "*Probability, Statistics, and Reliability for Civil and Environmental Engineers*," The Mc-Graw Hill Companies, Inc., New York.

Kreysig, E., 1983, "*Advanced Engineering Mathematics - Fifth Edition*," John Wiley & Sons, New York.

Lindfield, G. and J. Penny, 2000, "*Numerical Methods Using Matlab®,"* Prentice Hall, Upper Saddle River, New Jersey.

Magrab, E.B., S. Azarm, B. Balachandran, J. Duncan, K. Herold, and G. Walsh, 2000, "*An Engineer's Guide to MATLAB®*", Prentice Hall, Upper Saddle River, N.J., U.S.A.

McCuen, R.H., 1989,"*Hydrologic Analysis and Design - second edition*," Prentice Hall, Upper Saddle River, New Jersey.

Middleton, G.V., 2000, "*Data Analysis in the Earth Sciences Using Matlab®,"* Prentice Hall, Upper Saddle River, New Jersey.

Montgomery, D.C., G.C. Runger, and N.F. Hubele, 1998, "*Engineering Statistics*," John Wiley & Sons, Inc.

Newland, D.E., 1993, "*An Introduction to Random Vibrations, Spectral & Wavelet Analysis - Third Edition*," Longman Scientific and Technical, New York.

Nicols, G., 1995, "*Introduction to Nonlinear Science*," Cambridge University Press, Cambridge CB2 2RU, U.K.

Parker, T.S. and L.O. Chua, , "*Practical Numerical Algorithms for Chaotic Systems*," 1989, Springer-Verlag, New York.

Peitgen, H-O. and D. Saupe (editors), 1988, "*The Science of Fractal Images*," Springer-Verlag, New York.

Peitgen, H-O., H. Jürgens, and D. Saupe, 1992, "*Chaos and Fractals - New Frontiers of Science*," Springer-Verlag, New York.

Press, W.H., B.P. Flannery, S.A. Teukolsky, and W.T. Vetterling, 1989, "*Numerical Recipes - The Art of Scientific Computing (FORTRAN version)*," Cambridge University Press, Cambridge CB2 2RU, U.K.

Raghunath, H.M., 1985, "*Hydrology - Principles, Analysis and Design*," Wiley Eastern Limited, New Delhi, India.

Recktenwald, G., 2000, "*Numerical Methods with Matlab - Implementation and Application*," Prentice Hall, Upper Saddle River, N.J., U.S.A.

Rothenberg, R.I., 1991, "*Probability and Statistics*," Harcourt Brace Jovanovich College Outline Series, Harcourt Brace Jovanovich, Publishers, San Diego, CA.

Sagan, H., 1961,"*Boundary and Eigenvalue Problems in Mathematical Physics*," Dover Publications, Inc., New York.

Spanos, A., 1999,"*Probability Theory and Statistical Inference - Econometric Modeling with Observational Data*," Cambridge University Press, Cambridge CB2 2RU, U.K.

Spiegel, M. R., 1971 (second printing, 1999), "*Schaum's Outline of Theory and Problems of Advanced Mathematics for Engineers and Scientists*," Schaum's Outline Series, McGraw-Hill, New York.

Tanis, E.A., 1987, "*Statistics II - Estimation and Tests of Hypotheses*," Harcourt Brace Jovanovich College Outline Series, Harcourt Brace Jovanovich, Publishers, Fort Worth, TX.

Tinker, M. and R. Lambourne, 2000, "*Further Mathematics for the Physical Sciences*," John Wiley & Sons, LTD., Chichester, U.K.

Tolstov, G.P., 1962, "*Fourier Series*," (Translated from the Russian by R. A. Silverman), Dover Publications, New York.

Tveito, A. and R. Winther, 1998, "*Introduction to Partial Differential Equations - A Computational Approach*," Texts in Applied Mathematics 29, Springer, New York.

Urroz, G., 2000, "*Science and Engineering Mathematics with the HP 49 G - Volumes I & II*", www.greatunpublished.com, Charleston, S.C.

Urroz, G., 2001, "*Applied Engineering Mathematics with Maple*", www.greatunpublished.com, Charleston, S.C.

Winnick, J., , "*Chemical Engineering Thermodynamics - An Introduction to Thermodynamics for Undergraduate Engineering Students*," John Wiley & Sons, Inc., New York.

