MATHEMATICA TUTORIAL

Related Tutorials »

Second-Order PDEs

The general form of a linear second-order PDE is

$$a\frac{\partial^2 u}{\partial x^2}+b\frac{\partial^2 u}{\partial x\,\partial y}+c\frac{\partial^2 u}{\partial y^2}+d\frac{\partial u}{\partial x}+e\frac{\partial u}{\partial y}+f\,u=g.$$

Here $u=u(x, y)$, and a, b, c, d, e, f, and g are functions of x and y only—they do not depend on u. If $g=0$, the equation is said to be homogeneous.

The first three terms containing the second derivatives are called the *principal part* of the PDE. They determine the nature of the general solution to the equation. In fact, the coefficients of the principal part can be used to classify the PDE as follows.

The PDE is said to be *elliptic* if $b^2-4\,a\,c<0$. The Laplace equation has $a=1$, $b=0$, and $c=1$ and is therefore an elliptic PDE.

The PDE is said to be *hyperbolic* if $b^2-4\,a\,c>0$. The wave equation has $a=1$, $b=0$, and $c=-1$ and is therefore a hyperbolic PDE.

The PDE is said to be *parabolic* if $b^2-4\,a\,c=0$. The heat equation has $a=1$, $b=0$, and $c=0$ and is therefore a parabolic PDE.

`DSolve` can find the general solution for a restricted type of homogeneous linear second-order PDEs; namely, equations of the form

$$a\frac{\partial^2 u}{\partial x^2}+b\frac{\partial^2 u}{\partial x\,\partial y}+c\frac{\partial^2 u}{\partial y^2}=0.$$

Here a, b, and c are constants. Thus, `DSolve` assumes that the equation has constant coefficients and a vanishing non-principal part.

Following are some examples of the three basic types (elliptic, hyperbolic, and parabolic) and an explanation of their significance.

Here is the general solution for Laplace's equation, an elliptic PDE.

```
In[1]:= LaplaceEquation = D[u[x, y], {x, 2}] + D[u[x, y], {y, 2}] == 0;

In[2]:= DSolve[LaplaceEquation, u[x, y], {x, y}]

Out[2]= {{u[x, y] → C[1][i x + y] + C[2][-i x + y]}}
```

This general solution contains two arbitrary functions, `C[1]` and `C[2]`. The arguments of these functions, $y+i\,x$ and $y-i\,x$, indicate that the solution is constant along the imaginary straight line $y=-i\,x+\alpha$ when `C[2]=0` and along $y=i\,x+\alpha$ when `C[1]=0` . These straight lines are called characteristic curves of the PDE. In general, elliptic PDEs have imaginary characteristic curves.

Here is another elliptic PDE.

```
In[3]:= a = 3; b = 1; c = 5; b^2 - 4 a*c

Out[3]= -59

In[4]:= eqn = a*D[u[x, y], {x, 2}] + b*D[u[x, y], x, y] + c*D[u[x, y], {y, 2}] == 0;
```

Note the imaginary characteristic curves for the equation.

```
In[5]:= sol = DSolve[eqn, u, {x, y}]
```

$$\text{Out[5]= }\left\{\left\{u\to\text{Function}\left[\{x,y\},\ C[1]\left[\frac{1}{6}\left(-1+i\sqrt{59}\right)x+y\right]+C[2]\left[\frac{1}{6}\left(-1-i\sqrt{59}\right)x+y\right]\right]\right\}\right\}$$

The solution is verified as follows.

```
In[6]:= eqn /. sol // Simplify
```

Out[6]= {True}

This finds the general solution of the wave equation, a hyperbolic PDE. The constant c in the wave equation represents the speed of light and is set to 1 here for convenience.

```
In[7]:= WaveEquation = D[u[x, t], {x, 2}] - D[u[x, t], {t, 2}] == 0;
```

```
In[8]:= DSolve[WaveEquation, u[x, t], {t, x}]
```

```
Out[8]= {{u[x, t] → C[1][-t + x] + C[2][t + x]}}
```

The characteristic lines for the wave equation are $x = k + t$ and $x = k - t$ where k is an arbitrary constant. Hence the wave equation (or any hyperbolic PDE) has two families of real characteristic curves. If initial conditions are specified for the wave equation, the solution propagates along the characteristic lines. Also, any fixed pair of characteristic lines determine the null cone of an observer sitting at their intersection.

Here is another example of a hyperbolic PDE.

```
In[9]:= a = 2; b = 7; c = -1; b^2 - 4 a * c
```

Out[9]= 57

```
In[10]:= eqn = a * D[u[x, y], {x, 2}] + b * D[u[x, y], x, y] + c * D[u[x, y], {y, 2}] == 0;
```

Notice that the equation has two families of real characteristics.

```
In[11]:= sol = DSolve[eqn, u, {x, y}]
```

Out[11]= $\left\{\left\{u \to \text{Function}\left[\{x, y\}, C[1]\left[-\frac{1}{4}\left(7+\sqrt{57}\right) x + y\right] + C[2]\left[-\frac{1}{4}\left(7-\sqrt{57}\right) x + y\right]\right]\right\}\right\}$

The solution can be verified as follows.

```
In[12]:= eqn /. sol // Simplify
```

Out[12]= {True}

Finally, here is an example of a parabolic PDE.

```
In[13]:= a = 3; b = 30; c = 75; b^2 - 4 a * c
```

Out[13]= 0

```
In[14]:= eqn = a * D[u[x, y], {x, 2}] + b * D[u[x, y], x, y] + c * D[u[x, y], {y, 2}] == 0;
```

```
In[15]:= sol = DSolve[eqn, u, {x, y}]
```

```
Out[15]= {{u → Function[{x, y}, C[1][-5 x + y] + x C[2][-5 x + y]]}}
```

The equation has only one family of real characteristics, the lines $y = 5x + \alpha$. In fact, any parabolic PDE has only a single family of real characteristics.

The solution can be verified as follows.

```
In[16]:= eqn /. sol // Simplify
```

Out[16]= {True}

The heat equation is parabolic, but it is not considered here because it has a nonvanishing non-principal part, and the algorithm used by DSolve is not applicable in this case.

RELATED TUTORIALS

- Differential Equation Solving with DSolve

TUTORIAL COLLECTION

- Differential Equation Solving with DSolve
- Mathematics and Algorithms